



## Excel 公式实战练习 30 题

完美 Excel 出品  
微信公众号: **excelperfect**



练习才能熟能生巧，练习才能灵活运用。要掌握一项技能，唯有反复练习。对于 Excel 公式与函数来说，也是如此。

本电子书《Excel 公式实战练习 30 题》，汇集了作者在网上看到的、在书中找到的、在应用时或解答问题中使用的一些公式，并将公式进行详细解析，希望通过练习的方法，精进 Excel 技能。

在完美 Excel 微信公众号中发送消息：**公式实战**，即可获得本电子书文档下载链接和密码。

**坚持分享最好的Excel与VBA技术知识**

微信公众号：*excelperfect*

知嗒知识号：*完美Excel*







# 目录

练习 1. 获取当前单元格所在列的列字符 .....	1
题目 .....	1
公式思路 .....	1
公式解析 .....	1
小结 .....	2
练习 2. 在单元格列区域中输入连续的数字 .....	3
题目 .....	3
公式思路 .....	3
公式 .....	4
公式解析 .....	4
小结 .....	4
练习 3. 获取单元格区域中最长内容的单元格数据 .....	5
题目 .....	5
公式思路 .....	5
公式 .....	5
公式解析 .....	6
小结 .....	6
练习 4. 找出符合多列条件的值 .....	7
题目 .....	7
公式思路 .....	8
公式解析 .....	8
小结 .....	10

练习 5. 判断单元格区域中是否有重复值 .....	11
题目.....	11
公式思路.....	11
公式解析.....	12
小结.....	12
练习 6. 统计单元格区域中不重复值的数量 .....	13
题目.....	13
公式思路.....	13
公式解析.....	14
小结.....	16
练习 7. 获取单元格区域中的不重复值 .....	17
题目.....	17
公式思路.....	17
公式解析.....	18
小结.....	19
练习 8. 判断两个单元格区域是否有重复值 .....	21
题目.....	21
公式思路.....	21
公式解析.....	22
应用扩展.....	22
小结.....	24
练习 9. 比较两列的值并提取不相同的数据 .....	25
题目.....	25
公式.....	25
公式解析.....	26
扩展.....	27
小结.....	27
练习 10. 将表转换成单列 .....	29
题目.....	29
公式思路.....	29

公式.....	30
公式解析.....	30
小结.....	30
练习 11. 将矩形数据区域转换为一行或者一列 .....	31
题目.....	31
公式思路.....	32
公式解析.....	32
小结.....	34
练习 12. 将列数据转换为行数据 .....	35
题目.....	35
公式思路.....	36
公式解析.....	37
小结.....	37
练习 13. 求连续数据之和的最大值 .....	39
题目.....	39
公式思路.....	40
公式解析.....	40
小结.....	43
练习 14. 获取非空单元格 .....	45
题目.....	45
公式思路.....	45
公式.....	46
公式解析.....	46
练习 15. 获取单元格中字符串的最后一个单词 .....	49
题目.....	49
公式思路.....	49
公式解析.....	49
小结.....	53
练习 16. 统计单元格区域中无效数据数量 .....	55
题目.....	55

公式思路.....	56
公式解析.....	56
练习 17. 颠倒单元格区域中的数据 .....	59
题目.....	59
公式思路.....	59
公式解析.....	60
小结.....	61
练习 18. 求包含文本内容的单元格中的数字之和 .....	63
题目.....	63
公式思路.....	63
公式.....	63
公式解析.....	64
小结.....	64
练习 19. 提取字符串中的数字 .....	65
题目.....	65
公式思路.....	65
公式.....	65
公式解析.....	66
小结.....	66
练习 20. 获取每行中第一个非空单元格 .....	69
题目.....	69
公式思路.....	69
公式.....	70
公式解析.....	70
小结.....	71
练习 21. 从数据区域中提取满足多条件的值 .....	73
题目.....	73
公式.....	73
公式解析.....	74
小结.....	75



练习 22. 获取指定班级的学生姓名 .....	77
题目.....	77
公式.....	77
公式解析.....	78
小结.....	79
练习 23. 总是获取某列数值中的最后 5 个数值之和 .....	81
题目.....	81
公式.....	81
公式解析.....	82
小结.....	83
练习 24. 分别求不同班级大于 90 分的学生数 .....	85
题目.....	85
公式.....	86
公式解析.....	86
小结.....	86
练习 25. 对不同的班级分别排序 .....	87
题目.....	87
公式思路.....	88
公式.....	88
公式解析.....	88
小结.....	89
练习 26. 生成在 1 至 10 之间且没有重复值的随机数 .....	90
题目.....	90
公式.....	90
公式解析.....	91
小结.....	91
练习 27. 求指定日期在哪个季度 .....	92
题目.....	92
公式思路.....	92
公式.....	93

公式解析.....	93
小结.....	93
练习 28. 统计两个日期之间有多少个星期一在 1 日 .....	94
题目.....	94
公式.....	94
公式解析.....	94
小结.....	96
练习 29. 求 2018 年母亲节的日期 .....	98
题目.....	98
公式思路.....	98
公式解析.....	98
小结.....	99
练习 30. 使用数组公式创建日历 .....	100
题目.....	100
公式思路.....	101
公式解析.....	101
小结.....	102
关于完美 Excel .....	104
完美 Excel 微信公众号使用指南 .....	105

## 练习 1. 获取当前单元格所在列的列字符

### 题目

如何使用一个公式来获取当前单元格所在列的列字符呢？例如，当前单元格为 B2，那么公式的结果应该为其所在列的列字母 B。




图 1.1

### 公式思路

首先获取当前单元格的地址，然后查找地址中的列字符。

### 公式解析

获取当前单元格所在列的列字符的公式如下：

```
=LEFT (ADDRESS (ROW () , COLUMN () , 2) , FIND ("$", ADDRESS (ROW () , COLUMN () , 2)) - 1)
```

公式的结果如下图 1.2 所示：

B2		$\text{fx}$	<code>=LEFT (ADDRESS (ROW (), COLUMN (), 2), FIND ("\$", ADDRESS (ROW (), COLUMN (), 2))-1)</code>			
	A	B	C	D	E	
1						
2		B				

图 1.2

公式中，`ADDRESS (ROW (), COLUMN (), 2)` 获取当前单元格的地址，指定参数 2 表明为行绝对引用列相引用混合地址，即 `B$2`。

`FIND ("$", ADDRESS (ROW (), COLUMN (), 2))` 找到表示绝对引用的“\$”号在地址中的位置，即 2。将结果减 1，得到列字符在地址中的位置，即 1。

将上述获得的地址作为 `LEFT` 函数的第一个参数，列字符的位置作为第二个参数，即 `LEFT (B$2, 1)`，获得当前单元格的列字符为 B。

## 小结

巧妙地应用地址引用，获取所需结果。

有关 `ADDRESS` 函数的语法和应用，参见完美 Excel 微信公众号《Excel 函数学习 13：ADDRESS 函数》。

## 练习 2. 在单元格列区域中输入连续的数字

### 题目

使用公式在工作表任意单元格列区域中输入连续的数字，如下图 2.1 所示，在单元格区域 A2:A7 中输入 1~6。

要求：不能单独删除或修改该单元格区域中的任何一个数字；不能在该单元格区域中添加或删除行；在该区域最开始的单元格 A2 之前插入一行时，数字序号不变；将该区域复制到其他地方时，数字序号不变。

	A	B
1	序号	
2	1	
3	2	
4	3	
5	4	
6	5	
7	6	
8		

图 2.1

### 公式思路

很明显，要满足要求，需要使用数组公式。使用 ROW 函数来创建数字序列。

## 公式

选择单元格区域 A2:A7，输入数组公式：

```
=ROW(A2:A7)-ROW(A2)+1
```

记得输入完后按 Ctrl+Shift+Enter 键。

## 公式解析

公式中的：

```
ROW(A2:A7)
```

生成数组{2;3;4;5;6;7}。

然后，减去 ROW(A2) 即 2，得到数组{0;1;2;3;4;5}。

加上 1 后，得到结果数组{1;2;3;4;5;6}。

## 小结

- 虽然这个公式很简单，但它可以运用到一些大的更复杂的数组公式中，用来获取需要的数据。凭这一点，就值得我们好好体味！

## 练习 3. 获取单元格区域中最长内容的单元格数据

### 题目

如下图 3.1 所示的工作表，在单元格区域 A1:A7 中有一组数据，如何使用公式获取该区域中最长内容的单元格数据？即单元格 A2 包含的文本“excelperfect”。

	A	B	C	I
1	完美Excel		列A中最长的	
2	excelperfect		词语是：	
3	Access		excelperfect	
4	Word			
5	PowerPoint			
6	Excel			
7	Office			
8				

图 3.1

### 公式思路

先求出单元格区域中每个单元格中数据的长度，然后查找最长的长度所在的单元格位置，最后根据该位置获取相应单元格的值。

### 公式

在单元格 C3 中的数组公式如下：

```
=INDEX(A1:A7,MATCH(MAX(LEN(A1:A7)),LEN(A1:A7),0))
```

结果如图 3.2 所示。

	C3		$\{=INDEX(A1:A7,MATCH(MAX(LEN(A1:A7)),LEN(A1:A7),0))\}$					
	A	B	C	D	E	F	G	
1	完美Excel		列A中最长的					
2	excelperfect		词语是：					
3	Access		excelperfect					
4	Word		↑					
5	PowerPoint		数组公式：					
6	Excel		$=INDEX(A1:A7,MATCH(MAX(LEN(A1:A7)),LEN(A1:A7),0))$					
7	Office							
8								

图 3.2

## 公式解析

首先，公式中的：

**LEN(A1:A7)**

获取单元格区域 A1:A7 中各单元格数据的长度，即 {7;12;6;4;10;5;6}。

**MAX(LEN(A1:A7))**

获取长度值中的最大值，即 12。

因此，公式中的：

**MATCH(MAX(LEN(A1:A7)),LEN(A1:A7),0)**

变为：

**MATCH(12,{7;12;6;4;10;5;6},0)**

其结果为 2。

将该值代入 INDEX 函数，即公式变成：

**INDEX(A1:A7,2)**

得到单元格 A2 中的值。

## 小结

- LEN 函数可以用于生成一系列代表文本长度数据的数组。
- 经典的 MATCH 函数与 INDEX 函数相配合，获取单元格中的值。
- 也可以使用下面的数组公式得到同样的结果：

**=OFFSET(A1,MATCH(MAX(LEN(A1:A7)),LEN(A1:A7),0)-1,0,1,1)**

有兴趣的朋友可以自己思考一下。



## 练习 4. 找出符合多列条件的值

### 题目

工作表中有多列，其中有些列的值是相同的，我们要找到同时符合几列条件的值。例如下图 4.1 所示的工作表，列 A 中是学生姓名，有重复，列 B 中是科目，列 C 中是考试成绩。现在，我们需要找出某学生的某科目的成绩。

	A	B	C
1	学生姓名	科目	成绩
2	张三	语文	90
3	张三	数学	95
4	张三	英语	85
5	张三	物理	80
6	张三	化学	95
7	张三	生物	90
8	李四	语文	88
9	李四	数学	96
10	李四	英语	90
11	李四	物理	96
12	李四	化学	97
13	李四	生物	98
14	王五	语文	89
15	王五	数学	92

图 4.1

例如，要找出张三的物理成绩。在工作表中，一眼就可以看出是 80，但是如果学生多了呢？就不好找了！

如何使用公式找出张三的物理成绩？

## 公式思路

将列 A 和列 B 的值结合，用作供搜索的具有不重复值的列表；将张三和物理结合，用作搜索的字符串。找出行号后，其对应的列 C 中的值即为成绩。


## 公式解析

### 解法 1：使用辅助列+普通公式

在单元格 F2 中输入公式：

**=A2&B2**

下拉填充公式，得到供搜索的具有不重复值的列表，如下图 4.2 所示：



	A	B	C	D	E	F
1	学生姓名	科目	成绩			
2	张三	语文	90			张三语文
3	张三	数学	95			张三数学
4	张三	英语	85			张三英语
5	张三	物理	80			张三物理
6	张三	化学	95			张三化学
7	张三	生物	90			张三生物
8	李四	语文	88			李四语文
9	李四	数学	96			李四数学
10	李四	英语	90			李四英语
11	李四	物理	96			李四物理
12	李四	化学	97			李四化学
13	李四	生物	98			李四生物
14	王五	语文	89			王五语文
15	王五	数学	92			王五数学
16						

图 4.2

在单元格 H2 和 I2 中放置要查找的值，那么在单元格 H4 中输入查找公式为：

**=INDEX(C2:C15,MATCH(H2&I2,F2:F15,0))**

结果如下图 4.3 所示：

	A	B	C	D	E	F	G	H	I
1	学生姓名	科目	成绩						
2	张三	语文	90			张三语文		张三	物理
3	张三	数学	95			张三数学			
4	张三	英语	85			张三英语		80	
5	张三	物理	80			张三物理			
6	张三	化学	95			张三化学			
7	张三	生物	90			张三生物			
8	李四	语文	88			李四语文			
9	李四	数学	96			李四数学			
10	李四	英语	90			李四英语			
11	李四	物理	96			李四物理			
12	李四	化学	97			李四化学			
13	李四	生物	98			李四生物			
14	王五	语文	89			王五语文			
15	王五	数学	92			王五数学			
16									

图 4.3

MATCH 函数在单元格区域 F2:F15 中搜索由 H2 和 I2 组合的字符串，得到相匹配的值所在的行号，INDEX 函数在区域 C2:C15 中找到相应行的值。

### 解法 2：使用数组公式

在单元格 H4 中输入数组公式(公式输入完后要按 Ctrl+Shift+Enter 组合键):

**=INDEX(C:C,MATCH(H2&I2,\$A\$1:\$A\$15&\$B\$1:\$B\$15,0))**

结果如下图 4.4 所示:

.. ▾ {=INDEX(C:C,MATCH(H2&I2,\$A\$1:\$A\$15&\$B\$1:\$B\$15,0))}									
	A	B	C	D	E	F	G	H	I
1	学生姓名	科目	成绩						
2	张三	语文	90			张三语文		张三	物理
3	张三	数学	95			张三数学			
4	张三	英语	85			张三英语		80	
5	张三	物理	80			张三物理			
6	张三	化学	95			张三化学			
7	张三	生物	90			张三生物			
8	李四	语文	88			李四语文			
9	李四	数学	96			李四数学			
10	李四	英语	90			李四英语			
11	李四	物理	96			李四物理			
12	李四	化学	97			李四化学			
13	李四	生物	98			李四生物			
14	王五	语文	89			王五语文			
15	王五	数学	92			王五数学			
16									

图 4.4

公式中：**\$A\$1:\$A\$15&\$B\$1:\$B\$15** 将得到下面的数组：{"学生姓名科目";"张三语文";"张三数学";"张三英语";"张三物理";"张三化学";"张三生物";"李四语文";"李四数学";"李四英语";"李四物理";"李四化学";"李四生物";"王五语文";"王五数学"}，MATCH 函数在上述数组中查找 H2&I2 的值即"张三物理"，得到行号，再将其作为 INDEX 函数的参数在列 C 中找到符合条件的值 80。


## 小结

在编写公式时，往往不是一帆风顺的，总会有这样那样不满足基本函数的情形。然而，加上一些技巧，就可以化解这样的难题。本例中，列 A 中有重复值，但是将列 A 和列 B 合并后的列表就没有重复值了，接着将要查找的值合并后再查找，即可得到所需结果。

## 练习 5. 判断单元格区域中是否有重复值

### 题目

使用公式确定指定的单元格区域中是否有重复值。如下图 5.1 所示的工作表单元格区域 A1:A9，我们将其命名为 Data。



	A
1	1
2	5
3	6
4	8
5	5
6	1
7	6
8	1
9	8
10	

图 5.1

如果区域 Data 中有重复值，则返回 False；如果区域 Data 中的值都不一样，则返回 True。上图所示的区域 Data 中明显有多个重复值，因此应该返回 False。

### 公式思路


将区域 Data 中的每个值依次与区域中的所有值进行比较并统计与其相同的值的个数，由这些统计数组成一组数，如果这组数都由 1 组成，表明区域中的每个值都与其它值不同，即没有重复值，否则有重复值。

## 公式解析

使用下面的数组公式来判断单元格区域中是否有重复值：

**=MAX(COUNTIF(Data,Data))=1**

在单元格中输入公式后按 Ctrl+Shift+Enter 组合键，如下图 5.2 所示。



	A	B	C	D	E
1	1				
2	5		FALSE		
3	6				
4	8				
5	5				
6	1				
7	6				
8	1				
9	8				
10					

图 5.2

**COUNTIF(Data,Data)** 展开来就是：

COUNTIF({1;5;6;8;5;1;6;1;8},{1;5;6;8;5;1;6;1;8})。

首先测试 COUNTIF({1;5;6;8;5;1;6;1;8},1)，统计 1 在单元格区域 Data 中的次数，得到 3；接着测试 COUNTIF({1;5;6;8;5;1;6;1;8},5)，统计 5 在单元格区域 Data 中的次数，得到 2；..... 依此类推。最后，**COUNTIF(Data,Data)** 得到的结果是 {3;2;2;2;2;3;2;3;2}。这个结果作为 MAX 函数的参数，得到数组中的最大值为 3，不等于 1，因此最终的结果是 FALSE，表明单元格区域 Data 中有重复值。


## 小结

仔细体味这个公式，很有意思的是使用两个相同值作为参数进行比较统计。让我能感觉到 Excel 公式是多么的巧妙!!!

## 练习 6. 统计单元格区域中不重复值的数量

### 题目

统计单元格区域中有多少个不重复的值。如下图 6.1 所示的工作表：



Data	
	A
1	1
2	2
3	3
4	3
5	2
6	2

图 6.1

将单元格区域 A1:A6 命名为 Data，要使用公式求出区域 Data 中有多少个不重复的值。

因为数据很少，我们数一数，是 3 个，就是数字 1、2、3，但是如何用公式得出 3 呢？

### 公式思路

先计算出每个值在单元格区域中出现的次数，然后统计其出现频率，最后将频率值相加，即为不重复值的个数。

## 公式解析

在单元格中输入下面的数组公式：

**=SUM(1/COUNTIF(Data,Data))**

输入完后，记得按 Ctrl+Shift+Enter 组合键。结果如下图 6.2 所示：

C1			<b>f<sub>x</sub></b>			
	A	B	C	D	E	
1	1		3			
2	2					
3	3					
4	3					
5	2					
6	2					

图 6.2

公式中，**COUNTIF(Data,Data)** 统计单元格区域 Data 中每个值在区域中出现的次数。等价于 **COUNTIF({1;2;3;3;2;2},{1;2;3;3;2;2})**，首先使用 **COUNTIF({1;2;3;3;2;2},1)** 计算 1 在区域 Data 中出现的次数，得到结果 1；接着使用 **COUNTIF({1;2;3;3;2;2},2)** 计算 2 在区域 Data 中出现的次数，得到结果 3，.....，依此类推，最后得到的结果为{1;3;2;2;3;3}，即由区域中各个值在区域中出现的次数组成的数组。

**1/COUNTIF(Data,Data)** 计算所得数组{1;3;2;2;3;3}中每个值出现的频率，例如数组中第 2 个值 3 在出现的 3 次中占 1/3，即 0.333，因此，**1/COUNTIF(Data,Data)** 计算所得的结果为数组{1;0.333;0.5;0.5;0.333;0.333}。该数组作为 SUM 函数的参数，相加后的结果即为不重复值的数量（因为每个值在一组数中出现的频率之和为 1）。

下面，我们将求解过程分解，来进一步理解这个公式的原理。

在单元格 c1 中输入公式：

**=COUNTIF(Data,A1)**

并下拉至单元格 C6，统计区域 Data 中每个值出现的次数，结果如下图 6.3 所示。



C1	A	B	C	D
1	1		1	
2	2		3	
3	3		2	
4	3		2	
5	2		3	
6	2		3	

图 6.3

在单元格区域 D1:D6 中输入数组公式：

**=1/C1:C6**

得到每个值在区域 Data 中出现的频率。

D1	A	B	C	D
1	1		1	1
2	2		3	0.333333
3	3		2	0.5
4	3		2	0.5
5	2		3	0.333333
6	2		3	0.333333

图 6.4

对单元格区域 D1:D6 求和，即得到区域 Data 中不重复值的个数：

E1	A	B	C	D	E
1	1		1	1	3
2	2		3	0.333333	
3	3		2	0.5	
4	3		2	0.5	
5	2		3	0.333333	
6	2		3	0.333333	

图 6.5

注意，如果所求不重复值的区域中存在空单元格，会导致上述公式错误。

可以使用下面的数组公式解决：

**=SUM(IF(COUNTIF(Data,Data)=0,"",1/COUNTIF(Data,Data)))**

该公式巧妙地使用空格代替错误值#DIV/0!作为 SUM 函数的参数，将忽略掉空格而只求数值之和，最后得到所需结果。

## 小结


又一次惊叹公式的强大！其背后的原理，总是离不开基本的数学，好好体味这美妙的公式吧！

## 练习 7. 获取单元格区域中的不重复值

### 题目

在一个单元格区域中含有重复值，使用公式来获取该区域中的不重复值。

例如，下图 7.1 所示的工作表单元格区域 A1:A13，将其命名为 Data。在该区域中，含有很多重复值。现在要获取该区域中的不重复值。



	A
1	完美Excel
2	完美Excel
3	完美Excel
4	excelperfect
5	excelperfect
6	excelperfect
7	Office
8	Office
9	Office
10	Excel
11	Excel
12	Office
13	完美Excel

图 7.1

### 公式思路

首先求出单元格区域 Data 中每个值在区域中第 1 次出现的行号，然后根据行号

取出这些值。

## 公式解析

在单元格区域 C1:C13 中输入下面的数组公式：

```
=INDEX(Data, SMALL(IF(MATCH(Data, Data, 0)=ROW(INDIRECT("1:"&ROWS(Data)))), MATCH(Data, Data, 0), ""), ROW(INDIRECT("1:"&ROWS(Data)))))
```

按 Ctrl+Shift+Enter 组合键，即可得到区域 Data 中不重复值，如下图所示：

C1				
	A	B	C	D
1	完美Excel		完美Excel	
2	完美Excel		excelperfect	
3	完美Excel		Office	
4	excelperfect		Excel	
5	excelperfect		#NUM!	
6	excelperfect		#NUM!	
7	Office		#NUM!	
8	Office		#NUM!	
9	Office		#NUM!	
10	Excel		#NUM!	
11	Excel		#NUM!	
12	Office		#NUM!	
13	完美Excel		#NUM!	
14				

图 7.2

公式中，`MATCH(Data, Data, 0)` 得到数组 {1;1;1;4;4;4;7;7;7;10;10;7;1}，即区域 Data 中每个值在该区域中出现的行号。`ROW(INDIRECT("1:"&ROWS(Data)))` 得到数组 {1;2;3;4;5;6;7;8;9;10;11;12;13}，INDIRECT 函数将“1: 13”转换成行区域\$1:\$13。

`MATCH(Data, Data, 0)=ROW(INDIRECT("1:"&ROWS(Data)))` 即上述两个数组相比较，得到数组 {TRUE;FALSE;FALSE;TRUE;FALSE;FALSE;TRUE;FALSE;FALSE;TRUE;FALSE;FALSE;FALSE}，IF 函数根据该数组获取 `MATCH(Data, Data, 0)` 所得数组中的值，即 {1;"";"";4;"";"";7;"";"";10;"";"";""}，SMALL

函数分别取这个数组的第 1、2、3、...、13 最小值，即 {1;4;7;10;"";"";"";"";"";"";"";"";""}，将此数组作为 INDEX 函数的参数，分别取区域 Data 中对应的行的值。

从上图 7.2 所示的工作表中可以看出，对于数组公式中多余的单元格会显示 #NUM!。使用下面的数组公式避免显示 #NUM!。

```
=IFERROR(INDEX(Data, SMALL(IF(MATCH(Data, Data, 0)=ROW(INDIRECT("1:" & ROWS(Data))), MATCH(Data, Data, 0), ""), ROW(INDIRECT("1:" & ROWS(Data))))), "")
```

	A	B	C	D	E	F	G
1	完美Excel		完美Excel		完美Excel		
2	完美Excel		excelperfect		excelperfect		
3	完美Excel		Office		Office		
4	excelperfect		Excel		Excel		
5	excelperfect		#NUM!				
6	excelperfect		#NUM!				
7	Office		#NUM!				
8	Office		#NUM!				
9	Office		#NUM!				
10	Excel		#NUM!				
11	Excel		#NUM!				
12	Office		#NUM!				
13	完美Excel		#NUM!				

图 7.3

IFERROR 函数在错误值时输入空。

## 小结

- ROW 函数中不能再包括其它求值的函数，此时使用 INDIRECT 函数来间接引用。
- IFERROR 函数是 Excel 2007 及其后的版本中的函数，当第一个参数为错误值时，将另一个参数作为返回值。若要在 Excel 2013 中得到同

样的结果，则要将 IF 函数和 ISERR 函数结合使用。

- 公式中蕴含着一些通用思想，可以在其他类似情形中借鉴。

## 练习 8. 判断两个单元格区域是否有重复值

### 题目

如下图 8.1 所示的工作表，使用公式来判断单元格区域 A1:A3 与 C1:C3 中是否有重复值。

	A	B	C
1	完美Excel		Office
2	excelperfect		Excel
3	Excel		Word

图 8.1

从工作表中可以明显看出，这两个区域中都含有“Excel”，因此有重复值。

如何使用公式来判断呢？

### 公式思路

将单元格区域 A1:A3 中的每个值与 C1:C3 中的每个值比较，若都不相同，则没有重复值，否则就存在重复值。

## 公式解析

在单元格 E1 中输入下面的数组公式：

```
=OR (TRANSPOSE (A1:A3)=C1:C3)
```

其值为 TRUE，表明两个区域中存在重复值。如下图 8.2 所示。

E1					
	A	B	C	D	E
1	完美Excel		Office		TRUE
2	excelperfect		Excel		
3	Excel		Word		

图 8.2

公式中，使用 **TRANSPOSE 函数** 将 A1:A3 转置成水平方向的数组，即{"完美 Excel","excelperfect","Excel"}，然后，C1:C3 中的值 {"Office";"Excel";"Word"} 逐一与转置的数组相比较，即 C1 中的值 "Office" 与 {"完美 Excel","excelperfect","Excel"} 比较，C2 中的值 "Excel" 与 {"完美 Excel","excelperfect","Excel"} 比较，C3 中的值与 {"完美 Excel","excelperfect","Excel"} 比较，总共进行 9 次比较。若值相等，则返回 TRUE，否则为 FALSE。最后得到一个 3 行 3 列的数组：

```
{FALSE,FALSE,FALSE;FALSE,FALSE,TRUE;FALSE,FALSE,FALSE}
```

该数组作为 OR 函数的参数，获取比较后的结果。只要数组中有一个值为 TRUE，OR 函数的返回值就为 TRUE。本例中，数组中有一个值为 TRUE，因此返回 TRUE，则表明存在重复值。

## 应用扩展

### 扩展 1：要判断有无重复值的单元格区域方向不同

上文中判断重复值的单元格区域都在列中，然而如果要判断有无重复值的一个区域在列中，一个区域在行中，如下图 8.3 所示的工作表，判断单元格区域 A1:A3 与 C1:G1 是否有重复值，那么如何编写公式呢？

	A	B	C	D	E	F	G
1	完美Excel		Office	Word	Access	VBA	C
2	excelperfect						
3	Excel						



图 8.3

这种情形下，公式更简单！在单元格 C3 中输入数组公式：

```
=OR(A1:A3=C1:G1)
```

由于这两个区域没有重复值，因此其结果为 FALSE，如下图 8.4 所示。

C3								
	A	B	C	D	E	F	G	
1	完美Excel		Office	Word	Access	VBA	C	
2	excelperfect							
3	Excel		FALSE					
4								

图 8.4

单元格区域 A1:A3 中的每个值逐一与单元格区域 C1:G1 中的每个值相比较，共进行 15 次比较，得到数组：

```
{FALSE,FALSE,FALSE,FALSE,FALSE;FALSE,FALSE,FALSE,FALSE,FALSE;FALSE,FALSE,FALSE,FALSE,FALSE}
```

作为 OR 函数的参数。由于数组中全为 FALSE，因此结果为 FALSE，表明两个区域中没有重复值。

## 扩展 2：要比较的单元格区域具有更多的行列数

上述比较判断两个单元格区域是否具有重复值的方法，要求两个区域的方向不同，一个是逐行，一个是逐列。然而，对于 Excel 2003 来说，工作表列数为 256 列，对于 Excel 2007 及以上版本来说，工作表列数为 16384 列，若数据量超过这些列数，就不能使用上面的方法了。

此时，可以使用 MATCH 函数与 COUNT 函数组合来判断两个单元格区域是否有重复值。

例如，对于前面的例子，可以使用数组公式：

```
=COUNT(MATCH(A1:A3,C1:G1,0))
```

如果结果为 0，那么表明两个单元格区域没有重复值。如果结果为数字，那么该数字就表示这两个区域具有的重复值个数。

注：可以给要比较的两个区域命名，例如 Data1 和 Data2，然后使用名称代称公式里的区域，例如=COUNT(MATCH(Data1,Data2,0))，这样更灵活。

MATCH 函数在区域 Data2 中查找区域 Data1 中的值，若没有找到则产生错误值 #N/A，若找到则获取该值所在的列号。这样，生成由 #N/A 或数字组成的数组。将该数组作为 COUNT 函数的参数，统计数组中的数字。若没有数字，表明没有

重复值，则为 0。若有数字，则返回数字的个数，也即重复值的个数。

## 小结

- 要比较两个区域中的数据，这两个区域的方向应不同。若两个区域的方向相同，应使用 TRANSPOSE 函数将其中一个区域转置。TRANSPOSE 函数的用法详情参见[完美 Excel](#) 微信公众号中的文章《[Excel 函数学习 23：TRANSPOSE 函数](#)》。
- 上述方法比较的次数，也就是说产生的数组中元素个数取决于两个区域所占的行列数，例如第 1 个例子比较的次数为  $3 \times 3 = 9$  次，第 2 个例子比较的次数为  $3 \times 5 = 15$  次。
- Excel 工作表中列的数据代表数组中的每行，即每个数据后都是分号；而工作表中的行数据代表数组中的每列，即每个数据后都是逗号，最后才是分号。
- **COUNT 函数不会统计 #N/A 值，只返回非 #N/A 值的数量。**

## 练习 9. 比较两列的值并提取不相同的数据

### 题目

如下图 9.1 所示，要提取列 C 中西区超市有而列 A 中东区超市没有的水果，如何编写公式呢？

	A	B	C	D
1	东区超市		西区超市	
2	苹果		脐橙	
3	香梨		柚子	
4	猕猴桃		苹果	
5	柚子		香梨	
6			蜜桔	
7				
8				
9	提取西区超市中有而东区超市没有的水果			
10	脐橙			
11	蜜桔			
12				

图 9.1

### 公式

在单元格 A10 中的数组公式：

```
=IFERROR(INDEX($C$2:$C$6, SMALL(IF(ISNA(MATCH($C$2:$C$6, $A$2:$A$5, 0)), ROW($C$2:$C$6) - ROW($C$2) + 1), ROWS(A$10:A10))), "")
```

如图 9.2 所示。

A10		$\{=IFERROR(INDEX(\$C\$2:\$C\$6, SMALL(IF(ISNA(MATCH(\$C\$2:\$C\$6, \$A\$2:\$A\$5, 0)), ROW(\$C\$2:\$C\$6)-ROW(\$C\$2)+1), ROWS(A\$10:A10))), "")\}$		
	A	B	C	D
1	东区超市		西区超市	
2	苹果		脐橙	
3	香梨		柚子	
4	猕猴桃		苹果	
5	柚子		香梨	
6			蜜桔	
7				
8				
9	提取西区超市中有而东区超市没有的水果			
10	脐橙			
11	蜜桔			

图 9.2

向下拖至单元格中没有数据为止。

## 公式解析

公式中的：

**MATCH(\$C\$2:\$C\$6, \$A\$2:\$A\$5, 0)**

在单元格区域 A2:A5 中依次查找单元格区域 C2:C6 中的值，如果找到则返回该值在单元格区域 A2:A5 中的位置，否则返回 #N/A 值，最后生成的数组为 {#N/A; 4; 1; 2; #N/A}。将数组传递给 ISNA 函数，生成数组 {TRUE; FALSE; FALSE; FALSE; TRUE}，作为 IF 函数的第 1 个参数。

公式中的：

**ROW(\$C\$2:\$C\$6)-ROW(\$C\$2)+1**

生成数组 {1; 2; 3; 4; 5}，作为 IF 函数的第 2 个参数。

这样，IF 函数就生成数组 {1; FALSE; FALSE; FALSE; 5}，作为 SMALL 函数的第 1 个参数。

公式中的：

**ROWS (A\$10:A10)**

根据当前单元格生成一个数字，在 A10 中的数字为 1，在 A11 中为 2，将此数字作为 SMALL 函数的第 2 个参数。

公式演化为：

**=IFERROR (INDEX (\$C\$2:\$C\$6, SMALL ({1;FALSE;FALSE;FALSE;5}, 1) ), "")**

SMALL 函数求值后公式为：

**=IFERROR (INDEX (\$C\$2:\$C\$6, 1), "")**

获取单元格 C2 中的值。

## 扩展

如果要提取两列中相同的数据。如本例中，提取两个超市中都有的水果，数组公式为：

**=IFERROR (INDEX (\$C\$2:\$C\$6, SMALL (IF (ISNUMBER (MATCH (\$C\$2:\$C\$6, \$A\$2:\$A\$5, 0)), ROW (\$C\$2:\$C\$6) - ROW (\$C\$2) + 1), ROWS (E\$10:E10))), "")**

## 小结

- SMALL 函数忽略参数中的任何非数字的数据。
- ISNA 函数将错误值转换为 TRUE，否则为 FALSE。SNUMBER 函数的运用同理。
- IF 函数一次性判断获取数组。
- 公式适用于 Excel 2007 及以上版本。



## 练习 10. 将表转换成单列

### 题目

使用公式将如图 10.1 所示的工作表中的表区域 A2:C5 转换到列 E 中，如单元格区域 E1:E12。

	A	B	C	D	E	F
1	姓名	性别	班级		张三	
2	张三	男	301班		男	
3	李四	女	302班	转换成	301班	
4	王五	男	303班	→	李四	
5	赵六	女	304班		女	
6					302班	
7					王五	
8					男	
9					303班	
10					赵六	
11					女	
12					304班	
13						

图 10.1

### 公式思路

使用 INDEX 来提取数据，关键是要准确计算数据所在的行和列，这就需要找出行列数的规律并使用相应的运算得出这些行列数。

## 公式

选择单元格 E1 中，输入公式：

```
=INDEX($A$2:$C$5,INT((ROWS(E$1:E1)-1)/3)+1,MOD((ROWS(E$1:E1)-1),3)+1)
```

然后，拖动至单元格 E12，即可得到正确的结果。

## 公式解析

公式中的：

```
INT((ROWS(E$1:E1)-1)/3)+1
```

根据所在的单元格，生成相应的数字，例如在单元格 E1、E2、E3 中对应的数字是 1，在单元格 E4、E5、E6 中对应的数字是 2，在单元格 E7、E8、E9 中对应的数字是 3，在单元格 E10、E11、E12 中对应的数字是 4，作为 INDEX 函数的行参数，刚好对应着要提取的数据区域 A2:C5 中的第 2 至 5 行。

同样，公式中的：

```
MOD((ROWS(E$1:E1)-1),3)+1
```

根据所在的单元格，生成相应的数字，在单元格 E1 至 E12 中对应的数字依次为 1、2、3，1、2、3，1、2、3，1、2、3，即 1、2、3 重复出现，作为 INDEX 函数的列参数，刚好对应着提取的数据区域每行的 3 列。

## 小结

- 通过使用除法、求余等运算并结合 INT 函数，获取有规律的数字。



## 练习 11. 将矩形数据区域转换成一行或者一列

### 题目

将工作表中矩形数据区域转换成一行或者一列。例如将下图 11.1 所示的名为 Data 的区域：

	A	B	C	D	E
1					
2		5		6	7
3			99		11
4		333		888	999
5		1000	2000		3000
6					
7					
8					
9					

图 11.1

转换成单独的一列：

5
6
7
99
11
333
888
999
1000
2000
3000

图 11.2

或者单独的一行：

5	6	7	99	11	333	888	999	1000	2000	3000
---	---	---	----	----	-----	-----	-----	------	------	------

图 11.3

## 公式思路

获取数据所在的行号和列号，然后提取行列号所在单元格中的数据。

## 公式解析

### 按行顺序将区域中的数据转换成单独的一列

在单元格 B2 中输入公式：

```
=IF(INDIRECT(ADDRESS(ROW(Data)+INT((ROW()-ROW($B$2))/COLUMNS(Data)),COLUMN(Data)+MOD(ROW()-ROW($B$2),COLUMNS(Data)),4,, "Sheet1"))="", "",INDIRECT(ADDRESS(ROW(Data)+INT((ROW()-ROW($B$2))/COLUMNS(Data)),COLUMN(Data)+MOD(ROW()-ROW($B$2),COLUMNS(Data)),4,, "Sheet1")))
```

然后下拉至单元格 B17，如下图 11.4 所示。

<pre>=IF(INDIRECT(ADDRESS(ROW(Data)+INT((ROW()-ROW(\$B\$2))/COLUMNS(Data)),COLUMN(Data)+MOD(ROW()-ROW(\$B\$2),COLUMNS(Data)),4,, "Sheet1"))="", "",INDIRECT(ADDRESS(ROW(Data)+INT((ROW()-ROW(\$B\$2))/COLUMNS(Data)),COLUMN(Data)+MOD(ROW()-ROW(\$B\$2),COLUMNS(Data)),4,, "Sheet1")))</pre>					
逐行提取数据					
5					
6					
7					
99					
11					
333					
888					
999					
1000					
2000					
3000					

图 11.4

其中， $\text{ROW}(\text{Data}) + \text{INT}((\text{ROW}() - \text{ROW}(\$B\$2)) / \text{COLUMNS}(\text{Data}))$  获取 Data 区域中第 1 行第 1 列的数据所在的行。数学原理：用需要输入数据的区域中当前单元格所在行的行号减去首行行号的结果，除以被提取单元格区域 Data 的列数，所得结果加上 Data 区域起始行号，即为要在当前单元格中放置的数据在 Data 区域中的行号。

$\text{COLUMN}(\text{Data}) + \text{MOD}(\text{ROW}() - \text{ROW}(\$B\$2), \text{COLUMNS}(\text{Data}))$  获取 Data 区域中第 1 行第 1 列的数据所在的列。数学原理：用需要输入数据的区域中当前单元格所在行的行号减去首行行号的结果，除以被提取单元格区域 Data 的列数的余数，用所得结果加上 Data 区域起始列号，即为要在当前单元格中放置的数据在 Data 区域中的列号。

$\text{ADDRESS}(\text{行号}, \text{列号}, 4, , "Sheet1")$  以相对引用方式返回工作表 Sheet1 中相应行列号的地址。

$\text{INDIRECT}(\text{引用地址})$  返回引用地址所在单元格的数据。

使用 **IF 函数** 让结果为空的单元格仍然返回为空，而不是 0。

下面的几种情形与上文介绍的情形相似，原理相同，这里只给出公式，你可以自己调试体会。

### 按列顺序将区域中的数据转换成单独的一列

在单元格 B2 中的公式如下，然后拖动下拉至单元格 B17。

```
=IF(INDIRECT(ADDRESS(ROW(Data)+MOD(ROW()-ROW($B$2),ROWS(Data)),COLUMN(Data)+INT((ROW()-ROW($B$2))/ROWS(Data)),4,,"Sheet1"))="","",INDIRECT(ADDRESS(ROW(Data)+MOD(ROW()-ROW($B$2),ROWS(Data)),COLUMN(Data)+INT((ROW()-ROW($B$2))/ROWS(Data)),4,,"Sheet1")))
```

E	C	D	E	F
逐列提取数据				
5				
333				
1000				
99				
2000				
6				
888				
7				
11				
999				
3000				

图 11.5

### 按行顺序将区域中的数据转换成单独的一行

在单元格 B2 中的公式如下，然后向右拖动拉至单元格 Q2。

..		$=IF(INDIRECT(ADDRESS(ROW(Data)+INT((COLUMN()-COLUMN($B$2))/COLUMNS(Data)),COLUMN(Data)+MOD(COLUMN()-COLUMN($B$2),COLUMNS(Data)),4,,"Sheet1"))="","",INDIRECT(ADDRESS(ROW(Data)+INT((COLUMN()-COLUMN($B$2))/COLUMNS(Data)),COLUMN(Data)+MOD(COLUMN()-COLUMN($B$2),COLUMNS(Data)),4,,"Sheet1")))$						
	A	B	C	D	E	F	G	H
1	按行顺序转换为一行							
2		5		6	7		99	

图 11.6

### 按列顺序将区域中的数据转换成单独的一行

在单元格 B2 中的公式如下，然后向右拖动拉至单元格 Q2。

..		$=IF(INDIRECT(ADDRESS(ROW(Data)+MOD(COLUMN()-COLUMN($B$2),ROWS(Data)),COLUMN(Data)+INT((COLUMN()-COLUMN($B$2))/ROWS(Data)),4,,"Sheet1"))="","",INDIRECT(ADDRESS(ROW(Data)+MOD(COLUMN()-COLUMN($B$2),ROWS(Data)),COLUMN(Data)+INT((COLUMN()-COLUMN($B$2))/ROWS(Data)),4,,"Sheet1")))$						
	A	B	C	D	E	F	G	H
1	按列顺序转换为一行							
2		5	333	1000			99	

图 11.7

## 小结

本例的关键在于求出需要提取的数据在 Data 区域中的行列号，其中的运算是精华所在。可以先用代表行列的数字找到规律，再用相应求得行号列号的函数代替。

## 练习 12. 将列数据转换为行数据

### 题目


使用公式将工作表内一列中的重复项依次移到一行中。具体如下图 12.1 所示，要将工作表：



	A	B	C	D
1	姓名	资格证书		
2	张三	一级建造师		
3	张三	造价工程师		
4	张三	监理工程师		
5	李四	一级建筑师		
6	李四	设备工程师		
7	王五	电气工程师		
8				

图 12.1

转换成工作表：



	A	B	C	D	E
1	姓名	资格证书1	资格证书2	资格证书3	资格证书4
2	张三	一级建造师	造价工程师	监理工程师	
3	李四	一级建筑师	设备工程师		
4	王五	电气工程师			
5					
6					

图 12.2

可以看出，在“数据”工作表中，列 A 中有些姓名出现 1 次，有些出现 2 次，有些出现 3 次，但在列 B 中有不同的数据。

现在，需要将“数据”工作表中的数据转换成基于姓名的一行数据，即将与每个姓名相关的列 B 中的数据放置到单独的一行中。

使用数组公式来解决。在单元格 B2 中输入公式：

```
=IFERROR(INDEX(数据!$B$2:$B$7, SMALL(IF(数据!$A$2:$A$7=$A2, ROW(数据!$A$2:$A$7)-ROW(数据!$A$2)+1), COLUMNS($B2:B2))), "")
```

按 Ctrl+Shift+Enter 组合键完成输入。结果如下图 12.3 所示：

B2		{=IFERROR(INDEX(数据!\$B\$2:\$B\$7, SMALL(IF(数据!\$A\$2:\$A\$7=\$A2, ROW(数据!\$A\$2:\$A\$7)-ROW(数据!\$A\$2)+1), COLUMNS(\$B2:B2))), "")}				
	A	B	C	D	E	F
1	姓名	资格证书1	资格证书2	资格证书3	资格证书4	
2	张三	一级建造师				
3	李四					
4	王五					
5						

图 12.3

将单元格 B2 向右向下拖动，将公式复制到 B2:E4 区域后获得所需结果，如下图 12.4 所示：

	A	B	C	D	E
1	姓名	资格证书1	资格证书2	资格证书3	资格证书4
2	张三	一级建造师	造价工程师	监理工程师	
3	李四	一级建筑师	设备工程师		
4	王五	电气工程师			
5					
6					

图 12.4

## 公式思路

将列 A 中的值与“数据”工作表列 A 中的值相比较，如果相等则获取“数据”工作表中该值所在行的行号，然后根据获取的行号在“数据”工作表列 B 中找到相应的值并放置在本单元格中，如果没找到则为空。

## 公式解析

**数据!\$A\$2:\$A\$7=\$A2** 将单元格 A2 的值与“数据”工作表列 A 的值比较，因为前 3 行都相等，因此结果为：{TRUE;TRUE;TRUE;FALSE;FALSE;FALSE}。

**ROW(数据!\$A\$2:\$A\$7)-ROW(数据!\$A\$2)+1** 获取一个值为 1 至 6 的数组：{1;2;3;4;5;6}。

**IF(数据!\$A\$2:\$A\$7=\$A2,ROW(数据!\$A\$2:\$A\$7)-ROW(数据!\$A\$2)+1)** 的结果为数组{1;2;3;FALSE;FALSE;FALSE}。

**COLUMNS(\$B2:B2)** 返回当前单元格与 B2 单元格相间隔的列数，因为当前单元格就在 B2，因此返回 1。该数值用来作为 SMALL 函数的参数，获取第 k 个最小值，即需要第几行数据，从而获取列 A 值在“数据”工作表中对应的行号。本例中为 **SMALL({1;2;3;FALSE;FALSE;FALSE},1)**。返回第 1 个最小值，即 1。

然后，该值用作 INDEX 函数的参数：**INDEX(数据!\$B\$2:\$B\$7,1)**，获取“数据”工作表列 B 中相应行的数据，即“数据”工作表单元格 B2 中的数据。

IFERROR 函数用来在没有找到数据时给单元格填充空值，而不是显示错误值。

## 小结

- 个人觉得这个例子对于理解数组公式特别有用，值得反复练习与琢磨。
- 举一反三，本示例也可以用作查找重复值的参考方法。





## 练习 13. 求连续数据之和的最大值

### 题目

求连续 N 个数据中所有连续 M 个数据之和的最大值。

下图所示是完美 Excel 微信公众号文章《[Excel 函数学习 23: TRANSPOSE 函数](#)》的示例 3。

	A	B	C	D	E	F	G	H	I	J	K
1	连续几年中最高的总薪水										
2											
3											
4	年数	总薪水									
5	4	244,540									
6											
7	2008	2009	2010	2011	2012	2013	2014	2015	2016	2017	
8	54,993	59,718	66,383	56,254	62,185	51,572	58,733	53,249	57,340	69,406	
9											
10	237,348	244,540	236,394	228,744	225,739	220,894	238,728	-	-	-	Totals
11	1	1	1	1	1	1	1	0	0	0	Check
12											

图 13.1

在单元格 B5 中，给出了一个求连续几年薪水之和的最大值的公式。示例中是每连续 4 年（由单元格 A5 指定）薪水之和的最大值。

工作表的单元格 A10 中是 2008 年至 2011 年的薪水之和、B10 是 2009 年至 2012 年的薪水之和，.....，依此类推。其中单元格 B10 中的值就是所有连续 4 连薪水之和的最大值。

在工作表中，将单元格 A5 命名为“**Number**”。我们可以修改单元格 A5 中的年数，从而求出指定年数的薪水之和的最大值。

## 公式思路

先要获取连续的数据。如果是 4 年，那么有 7 组连续的数据。又因为总共有 10 个数据，因此形成一个 10 行×7 列的数组，在每列中需要计算的数据与该列列号对应，例如第 1 列是从第 1 个开始的 4 个数据，第 2 列是从第 2 个开始的 4 个数据。然后，将这些数据相加，形成一个有 7 个值的数组。最后，取出其最大值。

## 公式解析

单元格 B5 中的数组公式为：

```
=MAX(MMULT(A8:J8,--(ABS(TRANSPOSE(COLUMN(A8:J8))-COLUMN(OFFSET(A8:J8,0,0,1,COLUMNS(A8:J8)-Number+1))-(Number-1)/2)<Number/2)))
```

其中：

**COLUMN(A8:J8)** 的值为 {1,2,3,4,5,6,7,8,9,10}。

**TRANSPOSE(COLUMN(A8:J8))** 的值为 {1;2;3;4;5;6;7;8;9;10}。

也就是说，把 1 行 10 列的数组转置为 10 行 1 列的数组。

**COLUMNS(A8:J8)** 的值为 10，由于 Number 是单元格 A5 中的值，本例中为 4，因此 **COLUMNS(A8:J8)-Number+1** 的值为 10-4+1，即为 7。

**OFFSET(A8:J8,0,0,1,COLUMNS(A8:J8)-Number+1)** 即为

**OFFSET(A8:J8,0,0,1,7)**，值为 A8:G8。

**COLUMN(OFFSET(A8:J8,0,0,1,COLUMNS(A8:J8)-Number+1))** 即为

**COLUMN(A8:G8)**，值为 {1,2,3,4,5,6,7}。

这样，公式：

```
TRANSPOSE(COLUMN(A8:J8))-COLUMN(OFFSET(A8:J8,0,0,1,COLUMNS(A8:J8)-Number+1))
```

变为：

{1;2;3;4;5;6;7;8;9;10}-{1,2,3,4,5,6,7}

其结果为一个 10 行 7 列的数组：

{0,-1,-2,-3,-4,-5,-6;

1,0,-1,-2,-3,-4,-5;

2,1,0,-1,-2,-3,-4;

3,2,1,0,-1,-2,-3;

4,3,2,1,0,-1,-2;

5,4,3,2,1,0,-1;

6,5,4,3,2,1,0;

7,6,5,4,3,2,1;

8,7,6,5,4,3,2;

9,8,7,6,5,4,3}

此数组再减去  $(\text{Number}-1)/2$ ，本例中即  $(4-1)/2=1.5$ ，得到数组：

{-1.5,-2.5,-3.5,-4.5,-5.5,-6.5,-7.5;

-0.5,-1.5,-2.5,-3.5,-4.5,-5.5,-6.5;

0.5,-0.5,-1.5,-2.5,-3.5,-4.5,-5.5;

1.5,0.5,-0.5,-1.5,-2.5,-3.5,-4.5;

2.5,1.5,0.5,-0.5,-1.5,-2.5,-3.5;

3.5,2.5,1.5,0.5,-0.5,-1.5,-2.5;

4.5,3.5,2.5,1.5,0.5,-0.5,-1.5;

5.5,4.5,3.5,2.5,1.5,0.5,-0.5;

6.5,5.5,4.5,3.5,2.5,1.5,0.5;

7.5,6.5,5.5,4.5,3.5,2.5,1.5}

然后，使用 **ABS 函数**取上面的数组的绝对值，得到数组：

{1.5,2.5,3.5,4.5,5.5,6.5,7.5;

0.5,1.5,2.5,3.5,4.5,5.5,6.5;

0.5,0.5,1.5,2.5,3.5,4.5,5.5;

1.5,0.5,0.5,1.5,2.5,3.5,4.5;

2.5,1.5,0.5,0.5,1.5,2.5,3.5;

3.5,2.5,1.5,0.5,0.5,1.5,2.5;

4.5,3.5,2.5,1.5,0.5,0.5,1.5;

5.5,4.5,3.5,2.5,1.5,0.5,0.5;

6.5,5.5,4.5,3.5,2.5,1.5,0.5;

7.5,6.5,5.5,4.5,3.5,2.5,1.5}

将上面的数组与 **Number/2** 即  $4/2=2$  比较, 得到数组:

```
{True,False, False, False, False, False, False;  
True,True, False, False, False, False, False;  
True,True,True, False, False, False, False;  
True,True,True,True, False, False, False;  
False,True,True,True,True, False, False;  
False, False, True,True,True,True, False;  
False, False, False,True,True,True,True;  
False, False, False, False, True,True,True;  
False, False, False, False,False, True,True;  
False, False, False, False,False,False,True}
```

前面加上双减号 (**--**), 将 True 值转换为数字 1, False 值转换为 0。即得到数组:

```
{1,0,0,0,0,0,0;  
1,1, 0,0,0,0,0;  
1,1,1,0,0,0,0;  
1,1,1,1,0,0,0;  
0,1,1,1,1,0,0;  
0,0,1,1,1,1,0;  
0,0,0,1,1,1,1;  
0,0,0,0,1,1,1;  
0,0,0,0,0,1,1;  
0,0,0,0,0,0,1}
```

**MMULT (A8:J8, 上面的数组)** 将 **A8:J8** 形成的 1 行 10 列数组与上面获得的 10 行 7 列数组相乘得到一个 1 行 7 列的数组:

```
{237348,244540,236394,228744,225739,220894,238728}
```

最后, 使用 **MAX** 函数获取数组中的最大值。

下面, 来看看工作表第 11 行中的公式, 例如下面是 A11 中的公式:

**=-- (COLUMNS (A8:\$J\$8)>=Number)**

将当前单元格所在的列至列 J 的列数与 **Number** 值（本例中为 4）比较，若大于等于 **Number**，则为 True，并使用双减号将其转换为数字 1。

拖动 A11 至 J11。

下面，来看看工作表第 10 行中的公式，例如下面是 A10 中的公式：

**=IF(A11,SUM(A8:INDEX(A8:\$J\$8,1,Number)),0)**

如果对应的第 11 行的单元格（本例为 A11）中为 1，那么进行求和，**INDEX(A8:\$J\$8,1,Number)** 获取当前单元格之上对应的第 8 行中的单元偏移 **Number** 之后的单元格。然后与当前单元格之上对应的第 8 行中的单元格组成要求和的单元格区域，使用 **SUM** 函数求和。

拖到 A10 至 J10。

## 小结

- 获得这样一个矩阵数组真是不简单！
- 建议自己输入公式进行反复调试和体会，多练练手。



## 练习 14. 获取非空单元格

### 题目

尝试使用一个公式，来消除指定单元格区域中的空单元格，即获得的值中不包括空单元格，如下图 14.1 所示。

	A	B	C
1	苹果		苹果
2			香蕉
3	香蕉		
4			西瓜
5			
6			
7	西瓜		
8			
9			

图 14.1

### 公式思路

先找到非空单元格所在行的行号，获取行号并以行号作为 INDEX 函数的参数取出相应的值。

## 公式

选择单元格 C1:C7，输入公式：

```
=IFERROR(INDEX(A1:A7, SMALL(IF(A1:A7<>"", ROW(A1:A7)), ROW(A1:A7))), "")
```

按 Ctrl+Shift+Enter 组合键完成输入。

## 公式解析

下面，我们将公式分解，来看看是怎么一步一步得到答案的。

首先，找出非空单元格所在行的行号。选择单元格 C1:C7，输入公式：

```
=IF(A1:A7<>"", ROW(A1:A7))
```

按 Ctrl+Shift+Enter 组合键完成输入。结果如下图 14.2 所示：

	C1						
	A	B	C	D	E	F	G
1	苹果		1				
2			FALSE				
3	香蕉		3				
4			FALSE				
5		3	5				
6			FALSE				
7	西瓜		7				
8							

图 14.2

从图 14.2 中可以看出，公式将列 A 中的值与空值比较，不为空则在列 C 中相应的单元格输入非空单元格行号，而空单元格则输入 FALSE。

接下来，获取已经找出的非空单元格的行号。选择单元格 E1:E7，输入公式：

```
=SMALL(C1:C7, ROW(A1:A7))
```

按 Ctrl+Shift+Enter 组合键完成输入。结果如下图 14.3 所示：



E1	fx {=SMALL(C1:C7,ROW(A1:A7))}						
	A	B	C	D	E	F	G
1	苹果		1		1		
2			FALSE		3		
3	香蕉		3		5		
4			FALSE		7		
5		3	5		#NUM!		
6			FALSE		#NUM!		
7	西瓜		7		#NUM!		
8							

图 14.3

代表非空单元格行号的数值已依次输入到列 E 单元格中。ROW 函数得到一个数组 {1;2;3;4;5;6;7}，作为 SMALL 函数的参数，依次取出 C1:C7 中第 1 至第 7 小的值。

然后，将行号作为 INDEX 函数的参数取出值。选择单元格 G1:G7，输入公式：

**=INDEX(A1:A7,E1:E7)**

按 Ctrl+Shift+Enter 组合键完成输入。结果如下图 14.4 所示：

G1	fx {=INDEX(A1:A7,E1:E7)}							
	A	B	C	D	E	F	G	H
1	苹果		1		1		苹果	
2			FALSE		3		香蕉	
3	香蕉		3		5		3	
4			FALSE		7		西瓜	
5		3	5		#NUM!		#NUM!	
6			FALSE		#NUM!		#NUM!	
7	西瓜		7		#NUM!		#NUM!	
8								

图 14.4

可以看到，在列 G 中放置了非空单元格的值，但也放置了错误值。INDEX 函数依次取出列 A 中第 1、3、5、7 行的数据。

最后，使用 IFERROR 函数消除错误值。选择单元格 I1:I7，输入公式：

**=IFERROR(G1:G7,"")**

按 Ctrl+Shift+Enter 组合键完成输入。结果如下图 14.5 所示：

I1	fx {=IFERROR(G1:G7,"")}									
	A	B	C	D	E	F	G	H	I	J
1	苹果		1		1		苹果		苹果	
2			FALSE		3		香蕉		香蕉	
3	香蕉		3		5			3		3
4			FALSE		7		西瓜		西瓜	
5		3	5		#NUM!		#NUM!			
6			FALSE		#NUM!		#NUM!			
7	西瓜		7		#NUM!		#NUM!			
8										

图 14.5

如果是错误值，则为空。

将上述各步的公式组合，即可得到最终的公式。

## 练习 15. 获取单元格中字符串的最后一个单词

### 题目

使用公式来获取字符串的最后一个单词。如下图 15.1 所示，提取列 A 单元格中字符串的最后一个单词，将其放置到列 C 相应的单元格中。

	A	B	C
1	I am a student		student
2	A Dog		Dog
3			

图 15.1

### 公式思路

首先查找字符串中空格最后一次出现的位置，然后从该位置起提取字符串结尾部分的单词。

### 公式解析

在单元格 C1 中输入数组公式：

```
=RIGHT(A1,MATCH(" ",MID(A1,LEN(A1)-ROW(INDIRECT("1:" & LEN(A1)))+1,1),0)-1)
```

向下拖动至单元格 C2，结果如下图 15.2 所示。

C1			$\{=RIGHT(A1, MATCH(" ", MID(A1, LEN(A1)-ROW(INDIRECT("1:" & LEN(A1)))+1, 1), 0)-1)\}$	
	A	B	C	D
1	I am a student		student	
2	A Dog		Dog	
3				

图 15.2

为方便对数组公式的理解，先以单元格 A1 中的字符串为例，一步一步导出适用的公式。

第 1 步：如下图 15.3 所示，由于列 A 中的字符串共 14 个字符，因此在 B 列中选取单元格区域 B1:B14，输入数组公式：

```
=ROW(INDIRECT("1:" & LEN(A1)))
```

得到一个包含从 1 至 14 的连续数字的数组并将其放置在单元格区域中。

B1			$\{=ROW(INDIRECT("1:" & LEN(A1)))\}$	
	A	B	C	D
1	I am a student	1		
2		2		
3		3		
4		4		
5		5		
6		6		
7		7		
8		8		
9		9		
10		10		
11		11		
12		12		
13		13		
14		14		
15				

图 15.3

第 2 步：选择单元格区域 C1:C14，输入数组公式：

```
=LEN(A1)-B1:B14+1
```

得到与列 B 中数字逆序的数组并放置在单元格区域中。

C1 ▾ <span>fx</span> {=LEN(A1)-B1:B14+1}			
	A	B	C
1	I am a student	1	14
2		2	13
3		3	12
4		4	11
5		5	10
6		6	9
7		7	8
8		8	7
9		9	6
10		10	5
11		11	4
12		12	3
13		13	2
14		14	1
15			

图 15.4

第 3 步：选择单元格区域 D1:D14，输入数组公式：

```
=MID(A1,C1:C14,1)
```

使用 MID 函数按 C1:C14 中的数字从 A1 中的最后一个字符开始提取字符，并放置在单元格区域中。

D1 ▾ <span>fx</span> {=MID(A1,C1:C14,1)}				
	A	B	C	D
1	I am a student	1	14	t
2		2	13	n
3		3	12	e
4		4	11	d
5		5	10	u
6		6	9	t
7		7	8	s
8		8	7	
9		9	6	a
10		10	5	
11		11	4	m
12		12	3	a
13		13	2	
14		14	1	I
15				

图 15.5

第 4 步：在单元格 E1 中输入公式：

```
=MATCH(" ",D1:D14,0)
```

使用 MATCH 函数查找空格出现的位置。因为已经将 A1 中的字符串逆序排列，所以该位置就是原字符串中最后一个空格的位置。

E1    fx    =MATCH(" ",D1:D14,0)					
	A	B	C	D	E
1	I am a student	1	14 t		8
2		2	13 n		
3		3	12 e		
4		4	11 d		
5		5	10 u		
6		6	9 t		
7		7	8 s		
8		8	7		
9		9	6 a		
10		10	5		
11		11	4 m		
12		12	3 a		
13		13	2		
14		14	1 I		

图 15.6

第 5 步：在单元格 F1 中输入公式：

=LEN(A1)-E1+1

得到单元格 A1 字符串中最后一个空格的位置。


F1		=LEN(A1)-E1+1				
	A	B	C	D	E	F
1	I am a student	1	14 t		8	7
2		2	13 n			
3		3	12 e			
4		4	11 d			
5		5	10 u			
6		6	9 t			
7		7	8 s			
8		8	7			
9		9	6 a			
10		10	5			
11		11	4 m			
12		12	3 a			
13		13	2			
14		14	1 I			
15						

图 15.7

第 6 步：使用下面的公式：

```
=LEN(A1) - (LEN(A1) - E1 + 1)
```

得出字符串中最后一个单词的字符数，即字符串总的字符数减去最后一个空格在字符串中的位置的结果。

第 7 步：使用下面的公式：

```
=RIGHT(A1, LEN(A1) - (LEN(A1) - E1 + 1))
```

获取字符串中最后一个单词。

将最后得到的公式依次使用前一步得到的公式替换，最终得到只包含 A1 的公式，即本文开头的数组公式。

## 小结

- 本文提供了从字符串的结尾开始查找指定字符第一次出现的位置的一种方法。
- 在第 4 步中，其实已经得出了逆序排列时最后一个空格的位置，减去 1 即为最后一个单词的字符数。然后，将其直接作为 RIGHT 函数的参数，即可得到结果。
- 一步一步得出中间结果，然后再逐步替换得到最终的公式，这是编写大公式的一个通用方法。





## 练习 16. 统计单元格区域中无效数据数量

### 题目

如下图 16.1 所示的工作表，使用公式来统计单元格区域 C2:C6 中没有出现在单元格区域 A2:A9 中的数据数量。

	A	B	C	
1	设备编码库		设备编码	
2	SX001		SX002	
3	SX002		GD001	
4	SX003		SX002	
5	SX004		GD003	
6	SX005		SX006	
7	GD001			
8	GD002			
9	GD003			
10				

图 16.1

也就是说，单元格区域 C2:C6 中凡是没有出现在单元格区域 A2:A9 中的数据，都是无效数据。从工作表中可以明显看出，单元格区域 C2:C6 中的“SX006”没有出现在单元格区域 A2:A9 中，因此“SX006”为无效数据，即单元格区域 C2:C6 中的无效数据为 1。

如何使用公式来统计呢？

## 公式思路

在单元格区域 A2:A9 中查找单元格区域 C2:C6 中的每个值,统计没有找到的值的数量。

## 公式解析

在单元格 E1 中输入下面的数组公式:

```
=SUM(1*ISNA(MATCH(C2:C6,A2:A9,0)))
```

其值为 1,表明单元格区域 C2:C6 中没有出现在单元格区域 A2:A9 中的数据数为 1。如下图 16.2 所示。

E1						
	A	B	C	D	E	F
1	设备编码库		设备编码		1	
2	SX001		SX002			
3	SX002		GD001			
4	SX003		SX002			
5	SX004		GD003			
6	SX005		SX006			
7	GD001					
8	GD002					
9	GD003					
10						

图 16.2

公式中, MATCH 函数在单元格区域 A2:A9 中依次查找 C2:C6 中的值,返回各个值在 A2:A9 中的位置,如果没有找到则返回 #N/A,结果为 {2;6;2;8;#N/A},将其作为 ISNA 函数的参数,得到结果 {FALSE;FALSE;FALSE;FALSE;TRUE},然后与 1 相乘,将其转换为 {0;0;0;0;1},作为 SUM 函数的参数得到结果 1。

ISNA 函数检测一个值是否为 #N/A,返回 TRUE 或 FALSE。检测值可以是一个单元格、公式,或者是一个单元格、公式或数值的名称。

## 小结

- 使用 1 与布尔值相乘,将布尔值转换为 0 或 1。

- 本例具有一定的实用性，即可以用于判断输入的数据是否是数据库已经规定的合法数据。例如，代表设备编码库的单元格区域 A2:A9 是已经规定的合法数据，而单元格区域 C2:C6 是实际工作中输入的数据，那么可以用这个公式来判断实际输入的数据是否是已规定的合法数据。
- 可以使用名称来替换单元格区域，使公式更灵活。



## 练习 17. 颠倒单元格区域中的数据

### 题目

使用公式将单元格区域中的数据颠倒过来。例如，下图 17.1 所示工作表中的单元格区域 Data（即 A1:A7），使用公式将原来处于区域 Data 中第一个单元格 A1 中的数据放置到最后一个单元格，本例中为单元格 C7，将区域 Data 中第二个单元格 A2 中的数据放置到倒数第二个单元格 C6，.....，依此类推，直至将区域 Data 中最后一个单元格 A7 中的数据放置到第一个单元格 C1。



Data ▾		<i>f<sub>x</sub></i>	完美Excel
	A	B	C
1	完美Excel		Access
2	excelperfect		PowerPoint
3	Office		Word
4	Excel		Excel
5	Word		Office
6	PowerPoint		excelperfect
7	Access		完美Excel
8			
9			

图 17.1

### 公式思路

先获取单元格区域中最后一行的行号，提取该行号所在的数据放置在第一个单元格，然后获取区域中倒数第二行的行号，提取该行号所在的数据放置在第二个单元格，直至将区域中第一行所在单元格的数据放置到最后一个单元格。

## 公式解析

### 使用单个公式

在单元格 C1 输入公式：

```
=INDEX(Data,ROWS(Data)-ROW(A1)+1,1)
```

然后，下拉至单元格 C7。最后的结果如下图 17.2 所示：

C1	A	B	C	D	E
1	完美Excel		Access		
2	excelperfect		PowerPoint		
3	Office		Word		
4	Excel		Excel		
5	Word		Office		
6	PowerPoint		excelperfect		
7	Access		完美Excel		

图 17.2

公式的关键是： $ROWS(Data)-ROW(A1)+1$ ，其中  $ROWS(Data)$  返回单元格区域的行数，本例中为 7。 $ROW(A1)$  返回单元格所在行的行号，在单元格 C1 中返回 1，那么  $7-1+1$  的结果为 7；C2 中返回 2，那么  $7-2+1$  的结果为 6；.....；C7 中返回 7，那么  $7-7+1$  的结果为 1；因此  $ROWS(Data)-ROW(A1)+1$  的结果在 C1 至 C7 中依次为 7、6、5、4、3、2、1，分别作为 INDEX 函数的行参数，在区域 Data 中获取相应的值。

### 使用数组公式

选择单元格区域 C1:C7，输入数组公式：

```
=INDEX(Data,ROWS(Data)-ROW(INDIRECT("1:" & ROWS(Data)))+1,1)
```

结果如下图 17.3 所示：

	A	B	C	D
1	完美Excel		Access	
2	excelperfect		PowerPoint	
3	Office		Word	
4	Excel		Excel	
5	Word		Office	
6	PowerPoint		excelperfect	
7	Access		完美Excel	

图 17.3

其中, `ROW(INDIRECT("1:" & ROWS(Data)))` 生成数组 {1;2;3;4;5;6;7}, 与 `ROWS(Data)` 即 7 并且加 1 进行运算后, 生成数组 {7;6;5;4;3;2;1}, 将该数组作为 `INDEX` 函数的行参数, 依次从区域 `Data` 中取出值, 放置在相应的单元格中。

如果区域中有空单元格, 那么颠倒的相应单元格会显示 0。使用下面的数组公式, 将空单元格仍显示为空单元格:

```
=IF(INDEX(Data,ROWS(Data)-ROW(INDIRECT("1:" & ROWS(Data)))+1,1)="", "", INDEX(Data,ROWS(Data)-ROW(INDIRECT("1:" & ROWS(Data)))+1,1))
```

即使用 `IF` 函数判断获取的值是否为空, 如果为空则显示空, 否则显示数据。

## 小结

- 要想获取某区域中的值, 首先要考虑如何得到该值在该区域的行列号。这样, 就可以使用 `INDEX` 函数来获取相应的值。
- `ROW` 函数相当于编程语言中的变量, 根据单元格所在行的不同而得到不同的数值。
- `ROW` 函数配合 `INDIRECT` 函数使用, 不用硬编码, 更灵活。
- `INDEX` 函数的用法详见[完美 Excel](#) 微信公众号文章《Excel 函数学习 2: `INDEX` 函数》, `INDIRECT` 函数的用法详见[完美 Excel](#) 微信公众号文章《Excel 函数学习 14: `INDIRECT` 函数》。





## 练习 18. 求包含文本内容的单元格中的数字之和

### 题目

单元格中的数据包含文本和数字（如图 18.1），如何使用公式求出该单元格中的数字之和？

	A	B	C
1	abc123		6
2			

图 18.1

### 公式思路

先将文本屏蔽，然后求数字之和。

### 公式

数组公式如下：

```
=SUM(IFERROR(1*MID(A1,ROW(1:10),1),0))
```

## 公式解析

首先，公式：

```
MID(A1,ROW(1:10),1)
```

将单元格 A1 中的数据转换成一个数组：

{“a”;“b”;“c”;“1”;“2”;“3”;“”;“”;“”;“”}。

我们假设单元格 A1 中的数据长度不会超过 10 个。

然后，将数组乘以 1，即公式：

```
1*MID(A1,ROW(1:10),1)
```

得到数组{#VALUE!; #VALUE!; #VALUE!;1;2;3; #VALUE!; #VALUE!; #VALUE!; #VALUE!}。

接下来，使用 IFERROR 函数将错误值转换成 0，即公式：

```
IFERROR(1*MID(A1,ROW(1:10),1),0)
```

得到数组{0;0;0;1;2;3;0;0;0;0}。从而，将单元格中的数字分离了出来。

最后，SUM 函数对上面产生的数组求和，得到最终的结果。

## 小结

- 使用 MID 函数，可以实现数据分离。
- 数字与文本相乘将产生错误值。

## 练习 19. 提取字符串中的数字

### 题目

单元格中的数据包含文本和数字（如图 19.1），如何使用公式提取出该单元格中的数字？

	A	B	C
1	Excel2017		2017
2			
3			

图 19.1

### 公式思路

先找到字符串文本中第 1 个数字出现的位置，然后取出从该位置起的全部数据。

### 公式

数组公式如下：

```
=1*MID(A1,MATCH(FALSE,ISERROR(1*MID(A1,ROW(1:10),1)),0),255)
```

## 公式解析

首先，公式：

```
MID(A1,ROW(1:10),1)
```

将单元格 A1 中的数据转换成一个数组：

{“E”;“x”;“c”;“e”;“1”;“2”;“0”;“1”;“7”;“”}。

我们假设单元格 A1 中的数据长度不会超过 10 个。

然后，将数组乘以 1，即公式：

```
1*MID(A1,ROW(1:10),1)
```

得到数组{#VALUE!; #VALUE!; #VALUE!; #VALUE!; #VALUE!;2;0;1;7; #VALUE!}。

接下来，使用 IFERROR 函数，若数组中是错误值则转换成 TRUE，否则为 FALSE，即公式：

```
ISERROR(1*MID(A1,ROW(1:10),1))
```

得到数组{TRUE; TRUE; TRUE; TRUE; TRUE;FALSE; FALSE; FALSE; FALSE; TRUE}。

使用 MATCH 函数，查找数组中第一个 FALSE 出现的位置，即

```
MATCH(FALSE,ISERROR(1*MID(A1,ROW(1:10),1)),0)
```

得到值 6，即字符串中的第 6 个字符开始出现数字。此时的公式为  
=1\*MID(“Excel2017”,6,256)

即从字符串“Excel2017”的第 6 位开始提取 256 个字符数据，从而取出字符“2017”，然后将其乘以 1，转换成数字。

## 小结

- 使用 MID 函数，可以实现数据分离。
- 数字与文本相乘将产生错误值。
- MATCH 函数精确查找指定值第 1 次出现的位置。
- 数字与数字形式的文本相乘，将其转换为数字。





## 练习 20. 获取每行中第一个非空单元格

### 题目

如何使用公式获取每行中第一个非空单元格？例如下图 20.1 所示工作表，要求使用公式根据上半部分的表格求各项目的开工日期。

	A	B	C	D	E	F	G
1	日期	2017/8/6	2017/8/15	2017/8/28	2017/9/5	2017/9/10	2017/9/28
2	项目进度	地基勘探	勘探完成				
3			基础工程	基坑验收			
4					钢筋工程	钢筋验收	
5					模板工程	模板验收	
6						混凝土浇筑	混凝土验收
7							
8	项目	开工日期	完美Excel: 如何根据上表使用公式求出各项目的开工日期？				
9	地基勘探						
10	基础工程						
11	钢筋工程						
12	模板工程						
13	混凝土浇筑						
14							

图 20.1

也就是说，要求出单元格区域 B2:G6 的每行中第一个非空单元格对应的 B1:G1 中的日期。

### 公式思路

先找到该行中第 1 个非空单元格的位置，然后获取第 1 行 B1:G1 中该位置的值。

## 公式

在单元格 B9 中的数组公式如下：

```
=IF(COUNTA(B2:G2),INDEX($B$1:$G$1,MATCH(TRUE,B2:G2<>"",0)), "")
```

结果如图 20.2 所示。

B9	fx {=IF(COUNTA(B2:G2),INDEX(\$B\$1:\$G\$1,MATCH(TRUE,B2:G2<>"",0)), "")}						
	A	B	C	D	E	F	G
1	日期	2017/8/6	2017/8/15	2017/8/28	2017/9/5	2017/9/10	2017/9/28
2	项目进度	地基勘探	勘探完成				
3			基础工程	基坑验收			
4					钢筋工程	钢筋验收	
5					模板工程	模板验收	
6						混凝土浇筑	混凝土验收
7							
8	项目	开工日期					
9	地基勘探	2017/8/6	← 数组公式: =IF(COUNTA(B2:G2),INDEX(\$B\$1:\$G\$1,MATCH(TRUE,B2:G2<>"",0)), "")				
10	基础工程	2017/8/15	☆☆ 下拉复制				
11	钢筋工程	2017/9/5					
12	模板工程	2017/9/5					
13	混凝土浇筑	2017/9/10					
14							

图 20.2

## 公式解析

首先，公式：

```
MATCH(TRUE,B2:G2<>"",0)
```

找出第 2 行单元格区域 B2:G2 中第 1 个非空单元格的位置，本例中为 1。

然后，将该值代入 INDEX 函数，即变成公式：

```
INDEX($B$1:$G$1,1)
```

得到单元格 B1，即对应的第 1 行中 B1:G1 的值。

将该值代入后的公式：

```
=IF(COUNTA(B2:G2),B1,"")
```

然后，根据 COUNTA 函数的值来判断公式最后的值。

因为本例中 B2:G2 包含文本，因此最后的结果即为单元格 B1 中的值。

将该公式下拉至单元格 B13，即可求出各项目对应的开工时间。



## 小结

- 使用 MATCH 函数找到位置，使用 INDEX 函数获取相应位置的数据。
- 使用 <>"" 来生成代表空与非空单元格的数组。



## 练习 21. 从数据区域中提取满足多条件的值

### 题目

如下图 21.1 所示，单元格区域 A8:D18 中是销售数据，如何使用公式提取张三在 2017 年 5 月 1 日至 2017 年 12 月 1 日之间的销售数据？

	A	B	C	D
1	使用AND条件的数组公式			
2	条件区域	日期	日期	销售员
3		2017/5/1	2017/12/1	张三
7				
8	日期	产品	销售员	销量
9	2017/2/1	笔记本	张三	9
10	2017/5/6	手机	李四	25
11	2017/4/19	笔记本	王五	6
12	2017/7/9	台式机	张三	10
13	2017/8/10	传真机	李四	5
14	2017/6/6	笔记本	李四	11
15	2017/9/27	手机	王五	20
16	2017/8/25	台式机	李四	9
17	2017/11/5	传真机	张三	5
18	2017/10/28	台式机	王五	19
19				

图 21.1

### 公式

在单元格 F9 中的数组公式：

```
=IFERROR(INDEX(A$9:A$18, SMALL(IF($A$9:$A$18>=$B$3, IF($A$9:$A$18<=$C$3, IF($C$9:$C$18=$D$3, ROW($A$9:$A$18) - ROW($A$9) + 1))), ROWS(F$9:F9))), "")
```

如图 21.2 所示。

F9	{=IFERROR(INDEX(A\$9:A\$18, SMALL(IF(\$A\$9:\$A\$18>=\$B\$3, IF(\$A\$9:\$A\$18<=\$C\$3, IF(\$C\$9:\$C\$18=\$D\$3, ROW(\$A\$9:\$A\$18)-ROW(\$A\$9)+1))), ROWS(F\$9:F9))), "")}								
	A	B	C	D	E	F	G	H	I
1	使用AND条件的数组公式								
2	条件区域	日期	日期	销售员					
3		2017/5/1	2017/12/1	张三					
7									
8	日期	产品	销售员	销量		日期	产品	销售员	销量
9	2017/2/1	笔记本	张三	9		2017/7/9	台式机	张三	10
10	2017/5/6	手机	李四	25		2017/11/5	传真机	张三	5
11	2017/4/19	笔记本	王五	6					
12	2017/7/9	台式机	张三	10					
13	2017/8/10	传真机	李四	5					
14	2017/6/6	笔记本	李四	11					
15	2017/9/27	手机	王五	20					
16	2017/8/25	台式机	李四	9					
17	2017/11/5	传真机	张三	5					
18	2017/10/28	台式机	王五	19					

图 21.2

向右与向下拖至单元格中没有数据为止。

## 公式解析

公式中的：

**IF(\$A\$9:\$A\$18>=\$B\$3, IF(\$A\$9:\$A\$18<=\$C\$3, IF(\$C\$9:\$C\$18=\$D\$3,**

将单元格区域 A9:A18 中的数据与条件区域中的单元格 B3 和 C3 中的数据相比较，如果条件都满足，则将单元格区域 C9:C18 中的数据与单元格 D3 中的数据比较。

公式中的：

**ROW(\$A\$9:\$A\$18)-ROW(\$A\$9)+1**

生成数组{1;2;3;4;5;6;7;8;9;10}，作为 IF 函数的第 2 个参数。

这样，上述两个公式联合生成数组

{FALSE;FALSE;FALSE;4;FALSE;FALSE;FALSE;FALSE;9;FALSE}，作为 SMALL 函数的第 1 个参数。

公式中的：

**ROWS (F\$9:F9)**

根据当前单元格所在位置生成一个数字，在 F9 中的数字为 1，在 G9 中也为 1，在 F10 中为 2，将此数字作为 SMALL 函数的第 2 个参数。在刚生成的数组中取值。如果在单元格 F9 中，则得到的值为 4，即数据区域中的第 4 行，并将此作为 INDEX 函数的行参数。

在单元格 F9 中的公式演化为：

**=IFERROR (INDEX (A\$9:A\$18,4) , "")**

即获取到单元格 A12 中的值。

将单元格 F9 中的公式向右和向下拖动时，相对引用单元格自动调整，并获取相应单元格的值。

## 小结

- SMALL 函数忽略参数中的任何非数字的数据。
- 多个 IF 函数用于多个条件判断并获取数组。
- 公式适用于 Excel 2007 及以上版本。



## 练习 22. 获取指定班级的学生姓名

### 题目

如下图 22.1 所示，在单元格 D2 中指定班级名称，要获取数据区域 A1:B10 中该班级学生姓名，如何编写公式？

	A	B	C	D
1	学生姓名	班级		班级
2	李长江	301班		303班
3	谢小刚	303班		
4	张绿萍	302班		学生姓名
5	李来	303班		
6	潘十三	301班		
7	付水生	301班		
8	张三桥	303班		
9	彭天豪	302班		
10	杨笑天	303班		
11				

图 22.1

### 公式

在单元格 D5 中的数组公式：

```
=IFERROR(INDEX($A$2:$A$10, SMALL(IF($B$2:$B$10=D$2, ROW($A$2:$A$10)-ROW(A$2)+1), ROWS(G$4:G4))), "")
```

如图 22.2 所示。

D5	{=IFERROR(INDEX(\$A\$2:\$A\$10, SMALL( IF(\$B\$2:\$B\$10=D\$2, ROW(\$A\$2:\$A\$10)-ROW(A\$2)+1), ROWS(G\$4:G4))), "")}						
	A	B	C	D	E	F	G
1	学生姓名	班级		班级			
2	李长江	301班		303班			
3	谢小刚	303班					
4	张绿萍	302班		学生姓名			
5	李来	303班		谢小刚			
6	潘十三	301班		李来			
7	付水生	301班		张三桥			
8	张三桥	303班		杨笑天			
9	彭天豪	302班					
10	杨笑天	303班					
11							

图 22.2

向下拖至单元格中没有数据为止。

## 公式解析

公式中的：

**IF(\$B\$2:\$B\$10=D\$2**

将单元格区域 B2:B10 中的数据与单元格 D2 中的数据相比较，生成数组 {FALSE;TRUE;FALSE;TRUE;FALSE;FALSE;TRUE;FALSE;TRUE}。

公式中的：

**ROW(\$A\$2:\$A\$10)-ROW(A\$2)+1**

生成数组 {1;2;3;4;5;6;7;8;9}，作为 IF 函数的第 2 个参数。

这样，上述两个公式联合生成数组

{FALSE;2;FALSE;4;FALSE;FALSE;7;FALSE;9}，作为 SMALL 函数的第 1 个参数。

公式中的：

**ROWS(G\$4:G4)**

根据当前单元格所在位置生成一个数字，在 G4 中的数字为 1，在 G5 中为 2，将



此数字作为 SMALL 函数的第 2 个参数。

SMALL 函数产生的结果数值 INDEX 函数的行参数，在单元格区域 A2:A10 中获取相应的值。

将单元格 D5 中的公式向下拖动时，相对引用单元格自动调整，并获取相应单元格的值。

## 小结

- SMALL 函数忽略参数中的任何非数字的数据。
- IF 函数根据条件判断一次性生成相应的数组。
- 公式适用于 Excel 2007 及以上版本。



## 练习 23. 总是获取某列数值中的最后 5 个数值之和

### 题目

如下图 23.1 所示,在单元格区域 A1:A10 中有一系列数值,但有些单元格为空,使用公式求该区域最后 5 个数值之和,不计空格?

	A	B	C
1	1		12
2	2		
3	3		
4	1		
5			
6	5		
7	2		
8	1		
9			
10			
11			

图 23.1

### 公式

在单元格 C1 中的数组公式:

```
=SUM(INDEX($A$1:$A$10,LARGE(IF(ISNUMBER($A$1:$A$10),ROW($A$1:$A$10)-ROW($A$1)+1),5)):INDEX($A$1:$A$10,MATCH(9.99E+307,$A$1:$A$10)))
```

当单元格区域 A1:A10 中的数值改变时,单元格 C1 中的值随之更新,如下图 23.2 所示。

	A	B	C	
1	1		15	
2	2			
3	3			
4				
5				
6	5			
7	3			
8				
9	2			
10				
11				

图 23.2

## 公式解析

公式中的：

```
LARGE ( IF ( ISNUMBER ( $A$1 : $A$10 ) , ROW ( $A$1 : $A$10 ) - ROW ( $A$1 ) + 1 ) , 5 )
```

使用 IF 函数和 ISNUMBER 函数获取倒数第 5 个数值的位置。

其演算过程如下：

公式由

```
LARGE ( IF ( { TRUE ; TRUE ; TRUE ; FALSE ; FALSE ; TRUE ; TRUE ; FALSE ; TRUE ; FALSE } , { 1 ; 2 ; 3 ; 4 ; 5 ; 6 ; 7 ; 8 ; 9 ; 10 } ) , 5 )
```

变为

```
LARGE ( { 1 ; 2 ; 3 ; FALSE ; FALSE ; 6 ; 7 ; FALSE ; 9 ; FALSE } , 5 )
```

得到数字 2，即倒数第 5 个数值的位置。

代入 INDEX 函数中，得到单元格 A2。（与图 23.2 对应）

公式中的：

```
MATCH ( 9.99E+307 , $A$1 : $A$10 )
```

得到区域 A1:A10 中最后一个数值单元格的位置，与图 23.2 对应，其返回值应该是 9。

代入 INDEX 函数中，得到单元格 A9。

这样，对于图 23.2，最后 5 个数值所在的区域为 A2:A9。  
然后对其求和得到最终的结果。

## 小结

- 使用经典的 INDEX 函数和 MATCH 函数配合查找最后一个非空单元格。
- 体会公式中 ISNUMBER 函数、MATCH 函数的使用技巧。



## 练习 24. 分别求不同班级大于 90 分的学生数

### 题目

如下图 24.1 所示，在列 A 中是班级，列 B 中是各班级的成绩，要求出各班级大于 90 分的人数，即列 C 中得出的数字。

	A	B	C
1	优秀基本分：		90
2			
3	班级	成绩	大于优秀分的人数
4	301班	88	
5		92	
6		78	
7		95	2
8	302班	93	
9		91	
10		78	
11		90	2
12	303班	99	
13		89	
14		91	
15		95	3
16	304班	90	
17		86	
18		92	
19		88	1
20	平均分	89.6875	
21			

图 24.1

## 公式

在单元格 C4 中的公式：

```
=IF(A5<>"",COUNTIF(INDEX($B$3:B4,MATCH("做", $A$3:A4)):B4,">"&$C$1),"")
```

向下拖至单元格 C19。

## 公式解析

公式中的：

```
MATCH("做", $A$3:A4)
```

查找指定区域中的最后一个非空单元格的相对位置。由于公式向下拖动时，区域变化，因此求出的最后一个非空单元格位置也相应变化，以此作为 INDEX 函数的参数，求出动态区域中第 1 个单元格的位置。

然后，使用 COUNTIF 函数对求出的区域进行条件计数。

## 小结

- 使用经典的 INDEX 函数和 MATCH 函数配合以获得动态变化的区域。
- 体会公式中 MATCH 函数在查找最后一个非空单元格位置时使用的技巧。



## 练习 25. 对不同的班级分别排序

### 题目

如下图 25.1 所示的工作表，在单元格区域 A1:C9 中有一组学生成绩数据，如何使用公式分别对不同班级的学生根据成绩高低排序？

	A	B	C	D
1	姓名	班级	成绩	排序
2	张三	301	90	
3	李四	302	86	
4	王五	303	98	
5	赵六	303	92	
6	孙七	302	80	
7	钱八	302	85	
8	周九	301	95	
9	吴一	301	93	
10				

↓

	A	B	C	D
1	姓名	班级	成绩	排序
2	张三	301	90	3
3	李四	302	86	1
4	王五	303	98	1
5	赵六	303	92	2
6	孙七	302	80	3
7	钱八	302	85	2
8	周九	301	95	1
9	吴一	301	93	2
10				

图 25.1

## 公式思路

因为数据区域中要根据不同的班级分别排序该班级的成绩,所以需要区分不同的班级,并针对每个班级的学生成绩来排序。

## 公式

在单元格 D2 中的公式如下:

**=SUMPRODUCT(--(\$B\$2:\$B\$9=B2),--(\$C\$2:\$C\$9>C2))+1**

下拉至单元格 D9,结果如图 25.2 所示。

D2							
	A	B	C	D	E	F	G
1	姓名	班级	成绩	排序			
2	张三	301	90	3			
3	李四	302	86	1			
4	王五	303	98	1			
5	赵六	303	92	2			
6	孙七	302	80	3			
7	钱八	302	85	2			
8	周九	301	95	1			
9	吴一	301	93	2			
10							
11	单元格D2中的公式为:						
12	=SUMPRODUCT(--(\$B\$2:\$B\$9=B2),--(\$C\$2:\$C\$9>C2))+1						
13	** 向下拉复制到单元格D9						

图 25.2

## 公式解析

首先,公式中的:

**--(\$B\$2:\$B\$9=B2)**

将单元格区域 B2:B9 中的每个值与单元格 B2 中的值比较,得到数组 {TRUE;FALSE;FALSE;FALSE;FALSE;FALSE;TRUE;TRUE},双减号将逻辑值转换成{1;0;0;0;0;0;1;1}。

公式中的:

-- (\$C\$2:\$C\$9>C2)

将单元格区域 C2:C9 中的每个值与单元格 C2 中的值比较，得到数组 {FALSE;FALSE;TRUE;TRUE;FALSE;FALSE;TRUE;TRUE}，双减号将逻辑值转换成 {0;0;1;1;0;0;1;1}。

接着，SUMPRODUCT 函数将两个包括 0，1 的数组对应元素相乘后相加，得到对应的学生在所在班级的成绩排名。

即公式中的第 1 部分区分不同的班级，第 2 部分排序所在的成绩，结合起来即得到该学生所在班级的成绩排序。

## 小结

- SUMPRODUCT 函数的经典应用，多条件而无需数组公式。

## 练习 26. 生成在 1 至 10 之间且没有重复值的随机数

### 题目

使用公式生成位于两个值之间且没有重复值的随机数。如下图 26.1 所示，生成在 1 至 10 之间且没有重复值的随机数。

	A	B	C	D	E
1	生成在1至10之间并且没有重复值的随机数				
2		随机数			
3		8			
4		2			
5		6			
6		5			
7		10			
8		4			
9		3			
10		7			
11		9			
12		1			
13					

图 26.1

### 公式

在单元格 B3 中的数组公式：

```
=SMALL(IF(COUNTIF(B$2:B2,ROW($1:$10))<>1,ROW($1:$10)),1+INT(RAND()*(10-ROW()+ROW(B$3))))
```

拖至单元格 B12。

## 公式解析

下图 26.2 展示了公式中 SMALL 函数的第一个参数 IF (COUNTIF ()) 部分的运算过程。

	A	B	C	D	E	F	G	H	I	J
1										
2		随机数		公式 IF (COUNTIF (B\$2:B7,ROW(\$1:\$10))<>1,ROW(\$1:\$10)) 的原理						
3		6			ROW	COUNTIF	<>1	公式结果		
4		3	行1		1	1	FALSE	FALSE		
5		10	行2		2	0	TRUE	2		
6		9	行3		3	1	FALSE	FALSE		
7		1	行4		4	0	TRUE	4		
8			行5		5	0	TRUE	5		
9			行6		6	1	FALSE	FALSE		
10			行7		7	0	TRUE	7		
11			行8		8	0	TRUE	8		
12			行9		9	1	FALSE	FALSE		
13			行10		10	1	FALSE	FALSE		

图 26.2

单元格区域 E4:E13 中呈现的是 ROW (\$1:\$10) 的结果，即数字 1 到 10。

COUNTIF 函数统计列 E 中的数字在 B3:B7 中出现的次数并在 F4:F13 中放置统计的结果。

然后将列 F 中的结果与 1 比较，看是否等于 1，在 G4:G13 中放置比较后的结果。

如果列 G 中的值是 TRUE，表明列 E 中相应单元格的数值还没有被使用，将其放置到列 H 中，否则写入 FALSE。列 H 中的结果即为还没有被使用的数字。

作为 SMALL 函数第二个参数的公式：

**1+INT (RAND () \* (10-ROW () +ROW (B\$3)) )**

根据公式单元格所在的行，生成一个数字，该数字小于等于还没有被使用过的数字数。例如，如果公式所在的单元格为 B9，那么将生成 1 至 4 之间的一个整数作为 SMALL 函数的参数，提取剩下没有被使用的数字中的数。

## 小结

- SMALL 函数忽略参数中的任何非数字的数据。
- 公式创建了一组没有使用过的数字，然后选取这些数字中的一个作为随机数。
- 将 ROW 函数中的参数进行改变，生成所需区间的随机数。

## 练习 27. 求指定日期在哪个季度

### 题目

如图 27.1 所示的工作表，在单元格区域 A2:A5 中输入了日期，要求在列 B 相应的单元格中得出该日期所在的季度。例如，2017 年 8 月 9 日为第 3 季度，在单元格 B2 中的数字就为 3。

	A	B	C
1	日期	季度	
2	2017/2/15	1	
3	2017/8/9	3	
4	2017/5/20	2	
5	2017/11/11	4	
6			
7			

图 27.1

### 公式思路

先提取日期所在的月份，然后使用公式计算该月所在的季度数。

## 公式

选择单元格区域 B2:B5，输入数组公式：

```
=CEILING(MONTH(A2:A5),3)/3
```

输入完公式后，记得按 Ctrl+Shift+Enter 键。

## 公式解析

公式中的：

```
CEILING(MONTH(A2:A5),3)
```

由于 CEILING 函数返回的数字为向上舍入为最接近的指定基数的倍数，因此单元格 B2、B3、B4、B5 中对应返回的数分别为 3、9、6、12，除以 3 之后即为正确的季度数 1、3、2、4。

## 小结

- 灵活使用相关的运算函数进行运算，是编写所需公式的基础。
- 也可以不使用数组公式，将公式中的单元格区域换成对应的单元格。

## 练习 28. 统计两个日期之间有多少个星期一在 1 日

### 题目

使用公式求两个指定日期之间日期为 1 日的星期一数，如图 28.1 所示。

	A	B	C	D
1				
2		开始日期	2017/1/1	
3		结束日期	2017/12/31	
4		日期为1日的星期一数		
5				
6				

图 28.1

### 公式

在单元格 C2 中的公式：

```
=SUMPRODUCT(--(TEXT(ROW(INDIRECT(C2 & ":" & C3)),"aaaa d")="星期一 1"))
```

### 公式解析

公式中的：

```
ROW(INDIRECT(C2 & ":" & C3))
```

创建一组表示日期的序列号，如图 28.2 所示。



```
=SUMPRODUCT(--(TEXT({42736;42737;42738;42739;42740;
42741;42742;42743;42744;42745;42746;42747;42748;
42749;42750;42751;42752;42753;42754;42755;42756;
42757;42758;42759;42760;42761;42762;42763;42764;
42765;42766;42767;42768;42769;42770;42771;42772;
```

图 28.2

要了解 Excel 表示日期的原理，详见[完美 Excel 微信公众号文章《Excel 揭秘 5: 你看到的不一定是真实的—Excel 是怎样表示日期和时间的》](#)。INDIRECT 将数据转换成引用，并作为 ROW 函数的参数生成一系列日期序列号数字。

```
TEXT(ROW(INDIRECT(C2 & ":" & C3)),"aaaa d")
```

TEXT 函数将表示日期序列号数字格式成为“星期几 几号”的格式，即显示星期几和数字，若是星期一旦为 1 号则格式为“星期一 1”，如图 28.3 所示。

```
=SUMPRODUCT(--({ "星期日 1"; "星期一 2"; "星期二 3"; "星期三 4"; "星期四 5"; "星期五 6"; "星期六 7"; "星期日 8";
"星期一 9"; "星期二 10"; "星期三 11"; "星期四 12"; "星期五 13"; "星期六 14"; "星期日 15"; "星期一 16"; "星期二 17"; "星期三 18"; "星期四 19"; "星期五 20"; "星期六 21";
```

图 28.3

关于 TEXT 函数的详细使用详见[完美 Excel 微信公众号文章《Excel 函数学习 32: TEXT 函数》](#)，如何自定义数字格式请参考[完美 Excel 微信公众号文章《Excel 揭秘 2: 自定义数字格式原理与应用》](#)。

接着将 TEXT 函数格式化的数据与“星期一 1”比较，得到一个包含逻辑值 TRUE 和 FALSE 的数组，如图 28.4 所示。

```
=SUMPRODUCT(--({FALSE;FALSE;FALSE;FALSE;FALSE;
FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;
FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;
FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;
FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE;
```

图 28.4





## 练习 29. 求 2018 年母亲节的日期

### 题目

如何使用公式求出 2018 年母亲节的日期？

### 公式思路

母亲节 (Mother's Day)，是一个感谢母亲的节日，这个节日最早出现在古希腊；而现代的母亲节起源于美国，是每年 5 月的第二个星期日。因此，每年的母亲节的日期并不固定，但可以使用 5 月 1 日加上该月第二个星期日距 5 月 1 日的天数，来求出母亲节的日期。即：

母亲节日期 = 2018 年 5 月 1 日 + 第二个星期日距 5 月 1 日的天数

关键是，求出“第二个星期日距 5 月 1 日的天数”。

### 公式解析

我们先推导出求某月第 n 个星期几对应的日期的通用公式。

确定要求出的第 1 个星期几是小于还是大于该月的第 1 天对应的星期几。

如果要求出的第 1 个星期几小于该月第一天对应的星期几，则表达式为：

$7 - \text{WEEKDAY}(\text{DATE}(\text{该月第 1 天的日期})) + \text{代表要求出的第 1 个星期几的数值}$

如果要求出的第 1 个星期几大小该月第一天对应的星期几，则表达式为：

如果要求出第  $n$  个星期几，则还需要公式后面加上：

因此，通用公式为：

这样，求 2018 年母亲节日期的公式为：

结果为 2018 年 5 月 13 日，如下图 29.1 所示。

图 29.1

```
=DATE(2018,5,1)+1-WEEKDAY(DATE(2018,5,1))+(2-(1>=WEEKDAY(DATE(2018,5,1))))*7
```

## 小结

- 上面的两个公式虽然有差异，但原理相同，一个是以当月第 1 天为基准将日期向后推，一个是围绕当月第 1 天的日期向前向后推。
- 先找到一般规律，然后代入数字求结果。

## 练习 30. 使用数组公式创建日历

### 题目

如何使用数组公式创建如下图 30.1 所示的日历？

日历：2017年9月						
星期一	星期二	星期三	星期四	星期五	星期六	星期日
				9月1日	9月2日	9月3日
9月4日	9月5日	9月6日	9月7日	9月8日	9月9日	9月10日
9月11日	9月12日	9月13日	9月14日	9月15日	9月16日	9月17日
9月18日	9月19日	9月20日	9月21日	9月22日	9月23日	9月24日
9月25日	9月26日	9月27日	9月28日	9月29日	9月30日	

图 30.1

在单元格 B1 中输入日期时，显示该日期所在月的日历，如下图 30.2 所示。

	A	B	C	D	E	F	G	H
1	当月日期：	2015-5-15						
2	当月第一天所在星期几：	5						
3	当月第一天日期：	2015-5-1						
4								
5		星期一	星期二	星期三	星期四	星期五	星期六	星期日
6						5月1日	5月2日	5月3日
7		5月4日	5月5日	5月6日	5月7日	5月8日	5月9日	5月10日
8		5月11日	5月12日	5月13日	5月14日	5月15日	5月16日	5月17日
9		5月18日	5月19日	5月20日	5月21日	5月22日	5月23日	5月24日
10		5月25日	5月26日	5月27日	5月28日	5月29日	5月30日	5月31日
11								

图 30.2（完美 Excel 微信公众号中此图为动画演示图）

## 公式思路

一周有 7 天，一个月有 28 天、29 天、30 天、31 天，每个月的第一天开始的星期几不一样，如果分成含有 7 列的行的话，最少需要 4 行，最多需要 6 行。因此，需要创建 6 行×7 列的数组：

```
{1,2,3,4,5,6,7;8,9,10,11,12,13,14;15,16,17,18,19,20,21;22,23,24,25,26,27,28;29,30,31,32,33,34,35;36,37,38,39,40,41,42}。
```

然后，找到想要创建日历的月份的第 1 天，将代表这天的日期序号与数组相加，得到代表这个月的每天的序号，再减去该月份的第 1 天是当周的星期几，让日历正好能够与星期几相对应。

## 公式解析

获取当月日历的数组公式：

```
=IF(DAY(calendar)>15,IF(ROW()=6,"",calendar),IF(DAY(calendar)<15,IF(ROW()>9,"",calendar),calendar))
```

其中，calendar 是一个命名公式，即：

```
=week*7+weekday+B3-B2
```

其中，week 和 weekday 都是命名数组，即：

```
week:={0;1;2;3;4;5}
```

```
weekday:={1,2,3,4,5,6,7}
```

定义的名称如图 30.3。




图 30.3

`week*7+weekday` 得到数组:

{1,2,3,4,5,6,7;8,9,10,11,12,13,14;15,16,17,18,19,20,21;22,23,24,25,26,27,28;29,30,31,32,33,34,35;36,37,38,39,40,41,42}。

在上图 2 所示的工作表中,在单元格 B1 中放置了想要创建的日历月所在的日期。单元格 B3 中放置日期所在月份的第 1 天, B2 放置这一天是这一周的第几天。公式: `week*7+weekday+B3-B2` 得到从该月份第一天所在周的周一开始之后的 42 天的日期序列。

最后的数组公式中, `=IF(DAY(calendar)>15,IF(ROW()=6,"",calendar)` 表示如果上个月的日期大于 15 且在开始行,那么就显示为空,即不显示上月的日期;同样, `IF(DAY(calendar)<15,IF(ROW()>9,"",calendar)` 移除了下月的日期。

## 小结

- 创建日历看似复杂,但理解其背后蕴藏的道理后似乎很简单。
- 命名数组的应用技巧,值得借鉴。





## 关于完美 Excel

完美 Excel 是我创建的一个微信公众号，自 2017 年 5 月 15 日开始，每天推送一篇 Excel 与 VBA 技术和应用方面的文章。目前，共有 300 余篇实用文章可供广大 Excel 爱好者和使用者学习交流。这本电子书就是根据完美 Excel 上发表的《玩转 Excel 数据有效性》系列文章整理而成的。

每天早晨，完美 Excel 微信公众号：*excelperfect* 都会推送一篇关于 Excel 与 VBA 的相关文章。如果你有兴趣学习 Excel 和 VBA 的相关知识和实战技巧，可以关注完美 Excel 微信公众号，绝对不会让你失望！

可以通过下列方法关注[完美 Excel]微信公众号：

方法 1—在通讯录中搜索“完美 Excel”或者“*excelperfect*”后点击关注。

方法 2—扫一扫下面的二维码



## 完美 Excel 微信公众号使用指南

下图 1 为完美 Excel 微信公众号的界面。公众号名称显示在屏幕正上方，屏幕底部显示有“菜单栏”，目前设置的菜单为“技术精粹”、“VBA 精选”、“联系 me”。在底部左侧的小键盘图标为消息框入口，单击可进入消息框界面给完美 Excel 公众号发送消息。



图 1

下图 2、图 3、图 4 分别为底部 3 个菜单的子菜单。目前，菜单“技术精粹”中设置有“VBA 学习经验”、“玩转数据验证”、“快速学会 30 个函数”、“全部文章合集 1”等 4 个子菜单；菜单“VBA 精选”中设置有“最最基础入门”、“Range 对象详解”、“工作表对象详解”等 3 个子菜单；菜单“联系 me”中设置有“知识分享架构”、“个人声明”、“答疑解惑”、“坚持的美好”、“爱沐智养亲子中心”等 5 个子菜单。



图 2



图 3



图 4

单击这些子菜单会进入详细的文章页面或者文章整理的入口页面，方便读者浏览或查阅本公众号的文章。同时，这些子菜单会随着完美 Excel 微信公众号内容的增加而适时调整。

可以单击底部左侧的小键盘图标，进入发送消息界面，如图 5 所示。在文本框中输入想要发送的文字，单击底部的“发送”按钮，就可以将消息发送给完美 Excel 微信公众号。



图 5

大家应留意完美 Excel 微信公众号推送的文章中的一些信息，例如，我会在百度网盘中存放一些文档资料或者示例工作簿文件，并在文章中给出进入百度网盘下载的文本信息，你只需在发送消息框中输入我给出的文本，单击发送后，就会收到一条关于下载链接和密码的信息。单击链接并按提示输入密码后，即可获得相关的文档资料或示例工作簿文件了。

例如，在图 5 所示的界面中输入“Excel 动画图 2”后，会自动收到图 6 所示的信息，根据信息即可获取这个 Excel 动画图表文件。



图 6

希望大家在完美 Excel 微信公众号中能够学习到所需要的知识，获取到所需要的 Excel 应用技巧，提高自己的水平。